

Working Draft

Not to be Cited

Computing Biometrics at ICARDA

Volume II

M. Singh

Computer and Biometrics Services Unit



International Center for Agricultural Research in the Dry Areas
P. O. Box 5466, Aleppo, Syria


January 1995

Computing Biometrics at ICARDA

Volume II

M. Singh

Computer and Biometrics Services Unit



**International Center for Agricultural Research in the Dry Areas
P. O. Box 5466, Aleppo, Syria**

January 1995

Contents:


	Page
0. Introduction	1
1. Analysis of multi-locational trials conducted in incomplete blocks	2
2. Estimation and testing significance of genotypic and phenotypic correlations between traits using data from randomized complete block design.	8
3. Analysis of data from Diallel experiments-I	18
4. Comparing parallelism of non-linear curves-I	30
5. Comparing parallelism of non-linear curves-II	33
6. Analysis of long-term rotational trials: 1. Estimation of main effects and interactions.	36
7. Modelling germination in wheat- A robust method	50
8. Modelling absorbance due to rhizobia under salinity	55

0. Introduction

Biometric consultations often lead to the exploration of secrets of data using a number of computing biometric tools. During the course of scientist-biometrician interactive sessions, a number of computing biometric modules for specialized applications were sorted out. This volume contains the eight of such modules. For each module, we provide brief sections on introduction, areas of application, keywords, input/output, location of program, location of sample data and results, client(s), etc. Outputs running over large number of pages are not included in this document. However, they are available on a diskette.

Anyone interested in these programs is most welcome to ask for a copy of the programs and may also share them with other researchers, particularly working with National Agricultural Research Systems in West Asia and North Africa.

Seven of these programs are written in Genstat and one in FORTRAN, but volunteers are most welcome to code them in other computing languages as well.



Murari Singh
Sr. Biometrician

2 January 1995

Tel Hadya, Aleppo

**Computer and Biometrics Services Unit
ICARDA**

Serial Number -001

GENSTAT Modules in Computing Biometrics

Title: Analysis of multi-locational trials conducted in incomplete blocks

1. Introduction:

This program analyses data from multi-locational variety trials conducted in incomplete blocks. It displays individual location data analysis, combined analysis over locations to evaluate GxE interaction, computes stability indices, clustering of environments and of genotypes and estimates of variance components for genotypes, heritability estimates.

2. Areas of application and keywords:

Plant breeding and genetics, Physiology, Crop improvement.

Multi-locational varietal trials, combined analysis of variance, GxE interactions, stability indices, clustering, variance components, heritability.

3. Input/Output:

Input:

Plot-wise values of replications, genotypes, locations levels, yield variable(s).

Output:

Individual location ANOVA, means, se, %cv, combined analysis over locations, stability indices, clustering of environments and of genotypes, estimates of variance components for genotypes, heritability.

4. Location of the program:

Program:

A:\CompBiom\94\MLVT2.gen

(Software: GENSTAT 5 Rel 3.1)

5. Location of illustrative data and results:

Data:

A:\CompBiom\94\MLVT2.Txt

Results:

A:\CompBiom\94\MLVT2.opt

6. Client(s)

All plant breeders at ICARDA

7. Date

November 1994

8. Comments

M. Singh

Program:

```

*****
      GENSTAT program for analyzing multi-locational variety trials
      conducted in incomplete blocks.

```

```

      Software environment: Genstat 5 Rel 3.1

```

```

*****

```

```

OPEN 'MLVT2.Txt';CH=2;FI=IN
Scalar Nlines
Read[ch=2]Nlines
Skip[ch=2] Nlines

Scal NRep, NBlk,NGeno,NLoc
Read [ch=2] NLoc,NRep,NBlk,NGeno

Scal NObs, NGxNL
Calc NObs=NGeno*NLoc*NRep
Calc NGxNL=NGeno*NLoc

Unit [NObs]

Fact [Leve=NGeno] Geno
Fact [Leve=NLoc] Loc
Fact [Leve=NRep] Rep
Fact [Leve=NBlk] Blk
Read [Ch=2] Loc, Rep,Blk, Geno,Yield

```

```

" ===== Below is only for statistical programmers use ===== "

```

```

" 1. Individual locations analysis "

```

```

Vcomp[Fixed=Rep+Geno]Rep/Blk+Geno

```

```

Vari[Nval=NGeno] Mean[1...NLoc], GenoMean

```

```

For l=1...NLoc      ; MN=Mean[1...NLoc]
Rest Yield; Cond=Loc.EQ.l
Reml[prin=m,c,means,s,w] Yield
Vkeep Geno; Means=TDum
Equa TDum;MN
Dele[Rede=Y] TDum
Rest Yield
Endf

```

Varf[Nval=NLoc] LocMean

" 2. Combined analysis over locations "

Vcomp[Fixed=Rep+Geno*Loc]Loc+Loc.Rep/Blk+Geno/Loc
Reml[prin=m,c,means,s,w] Yield
Vkeep Geno; Means=TDum
Equa TDum; GenoMean
Dele[Rede=Y] TDum

Vkeep Loc; Means=TDum
Equa TDum; LocMean
Dele[Rede=Y] TDum

Fact[Leve=NGeno; Valu=1...NGeno] GenoNum
Prin GenoNum,Mean[1...NLoc],GenoMean; field=7
Prin[orie=a] LocMean ;fiel=7

" 3. Compute stability indices "

Varf[nvalu=NGeno] GenoCV
Calc GenoCV=100.*Sqrt(Vvar(IP(Mean[1...NLoc])))/GenoMean

Matr [Rows=NGeno; Colu=NLoc] GE : & [Rows=NLoc; Colu=NGeno] EG
Equa IP(Mean[1...NLoc]) ; EG
Calc GE=Tran(EG)

Varf[Nval=NLoc] GMean[1...NGeno]
Equa GE; IP(GMean[1...NGeno])

Dele GE, EG

Varf[Nval=NGxNL] GEData
Equa IP(Mean[1...NLoc]); GEData

Varf[Nval=NGeno] Slope,DeviMs, Wricke,Pla_Pet,Plaisted, Shukla

For l=1...NGeno ; Y=GMean[1...NGeno]
Model Y ; Fitt=F
Fit[prin=*] LocMean
RKeep ; Est=Est ; Devi=SS ; DF=df
Calc Slope\$(l) =Est \$(2) : & DeviMs \$(l) =SS/df

"

Graph Y,F; LocMean; symb='o','.'; Meth=p,c

"

Endf

Fact[Leve=NGeno; Nval=NGxNL] Geno1
 Fact[Leve=NLoc; Nval=NGxNL] Loc1
 Vari[Nval=NGxNL] GEInt

Gene Loc1, Geno1
 Bloc
 Trea Loc1+Geno1
 Anov GEData; Res=GEInt

Dele GEData

Calc GEInt=GEInt*GEInt
 Tabu [Class=Geno1] GEInt; Tota=TDum

Equa TDum; Wricke
 Dele[Rede=Y] TDum

Scal SsGE
 Calc SsGE=Sum(Wricke)

Calc Pla_Pet=(NGeno*Wricke+SsGE)/(2*(NGeno-1)*(NLoc-1))
 Calc Plaisted=(-NGeno*Wricke/(NGeno-1)+SsGE)/((NGeno-2)*(NLoc-1))
 Calc Shukla = (NGeno*Wricke - SsGE/(NGeno-1))/(NGeno-2)/(NLoc-1)

" Correlation between indices "

Corr[Print=c] GenoMean, Slope, DeviMs,GenoCV,Wricke,Pla_Pet,Plaisted,Shukla
 Print GenoNum, GenoMean, Slope, DeviMs,GenoCV,Wricke,Pla_Pet,Plaisted,Shukla

Graph Slope; GenoMean; Symb=GenoNum
 Graph GenoCV; GenoMean; Symb=GenoNum
 Graph DeviMs; GenoMean; Symb=GenoNum
 Graph Wricke; GenoMean; Symb=GenoNum

Vari[nval=NGeno]RGenoMn,RSlope,RDeviMs,RGenoCV,RWricke,RPla_Pet,RPlaist,RShukla
 Vari[values=1...NGeno]Order0,Order1
 For D= GenoMean, Slope, DeviMs,GenoCV,Wricke,Pla_Pet,Plaisted,Shukla ; \
 DD= RGenoMn, RSlope, RDeviMs,RGenoCV,RWricke,RPla_Pet,RPlaist,RShukla
 Sort[dire=D; Index=D] Order0
 Sort Order0,Order1
 Calc DD=Order1
 endf

" Correlations between ranks "

Corr[Prin=c]RGenoMn, RSlope, RDeviMs,RGenoCV,RWricke,RPla_Pet,RPlaist,RShukla

" 4. Clustering of Environments "

Symm[Rows=NLoc] Simi

```
Fsim[Simi=Simi] GMean[1...NGeno]; Test=Euclidean
Hclu[ Prin=a,d] Simi
Dele[Rede=Y] Simi
```

" 5. Clustering of Genotypes "

```
Symm[Rows=NGeno] Simi
Fsim[Simi=Simi] Mean[1...NLoc]; Test=Euclidean
Hclu[ Prin=a,d] Simi
Dele[Rede=Y] Simi
```

" 6. Estimation of variance components"

```
Vcom[Fix=Loc] Loc+Loc.Rep/Blk+Geno+Loc.Geno
REML Yield
```

" 7. Estimation of heritabilities"

" 7.1 From individual environments"

```
SCAL SGg2,SGe2,h2
Scal Vgg,Vge,Vee,Bias, Seh2
symm[3] Vcov
symm[2] Vcov_r
```

For i=1...NLoc

Rest Yield; Loc.eq.i

" Using incomplete block design: block effects random "

```
VCOMP[fixed=Rep] RANDOM=Rep/Blk+Geno
REML[print='] Yield
```

```
VKEEP[SIGMA2=SGe2;vcov=Vcov] Geno; COMP=SGg2
CALC Vgg,Vge,Vee=Vcov$(2,3,3; 2,2,3]
CALC h2=SGg2/(SGg2+SGe2)
CALC One_h22=(1-h2)**2
CALC Bias=One_h22*((1-h2)*Vgg-h2*Vge)/(h2*SGe2*SGe2)
CALC Seh2=(1-h2)*SQRT(One_h22*Vgg-2*h2*(1-h2)*Vge+Vee*h2**2)/SGe2
```

```
PRINT SGg2,SGe2,h2,Bias,Seh2
```

" Under complete blocks "

```
VCOMP[fixed=Rep] RANDOM=Rep+Geno
REML[print='] Yield
```

```
VKEEP[SIGMA2=SGe2;vcov=Vcov_r] Geno; COMP=SGg2
EQUA Vcov_r ; lp(Vgg,Vge,Vee)
CALC h2=SGg2/(SGg2+SGe2)
```

```

CALC One_h22=(1-h2)**2
CALC Bias=One_h22*((1-h2)*Vgg-h2*Vge)/(h2*SGe2*SGe2)
CALC Seh2=(1-h2)*SQRT(One_h22*Vgg-2*h2*(1-h2)*Vge+Vee*h2**2)/SGe2

```

```
PRINT SGg2,SGe2,h2,Bias,Seh2
```

```
rest Yield
Endf
```

" 8.2 Overall environments "

```

SCALAR SGg2,SGe2,SGi2,SGb2,Vgg,Vee,Vii,Vgi,Vge,Vie, h2, Bias,Se
SYMM[4] Vcov
Symm[3] Vcov_r

```

" Under incomplete blocks: Block effects random "

```

VCOMP[abso=Loc; Fixed=Loc/Rep] RANDOM=Loc+Loc.Rep+Loc.Rep.Blk+Geno+Geno.Loc
REML[print='] Yield

```

```

VKEEP[SIGMA2=SGe2;vcov=Vcov] Geno+Geno.Loc; COMP=SGg2,SGi2
CALC Vgg,Vgi,Vii,Vge,Vie,Vee=Vcov$[2,3,3,4,4,2,2,3,2,3,4]
CALC h2=SGg2/(SGg2+SGi2+SGe2)
CALC Bias =h2*(Vgg-h2*(Vgg+Vgi+Vge))/SGg2/SGg2
CALC Se=Vgg+Vii+Vee+2*(Vgi+Vge+Vie)
CALC Se=Vgg+h2*h2*Se-2*h2*(Vgg+Vgi+Vge)
CALC Se=h2*SQRT(Se)/SGg2

```

```
Print h2, Bias, Se
```

" Under complete blocks "

```

VCOMP[abso=Loc; Fixed=Loc/Rep] RANDOM=Loc+Loc.Rep+Geno+Geno.Loc
REML[print='] Yield

```

```

VKEEP[SIGMA2=SGe2;vcov=Vcov_r] Geno+Geno.Loc; COMP=SGg2,SGi2
EQUA Vcov_r ; lp(Vgg,Vgi,Vii,Vge,Vie,Vee)
CALC h2=SGg2/(SGg2+SGi2+SGe2)
CALC Bias =h2*(Vgg-h2*(Vgg+Vgi+Vge))/SGg2/SGg2
CALC Se=Vgg+Vii+Vee+2*(Vgi+Vge+Vie)
CALC Se=Vgg+h2*h2*Se-2*h2*(Vgg+Vgi+Vge)
CALC Se=h2*SQRT(Se)/SGg2

```

```
Print h2, Bias, Se
```

```

Clos
Stop

```

Output: Not included

**Computer and Biometrics Services Unit
ICARDA**

Serial Number -002

FORTTRAN Modules in Computing Biometrics

Title: **Estimation and testing significance of genotypic and phenotypic correlations between traits using data from randomized complete block design.**

1. Introduction:

This program performs bivariate analysis of variance to compute genotypic, phenotypic and environmental correlations between traits observed using a randomized complete block design. For details, see Singh, M. 1992. Genotypic and Phenotypic correlations in Plant Traits. ICARDA-033:May 1992. International Center for Agricultural research in the Dry Areas, P.O. Box 5466, Aleppo, Syria. pp 279.

2. Areas of application and keywords:

Plant breeding and genetics, Crop improvement.

3. Input/Output:

Input:

Plot-wise values of replications (i.e. complete blocks), genotype, variable(s).

Output:

Bivariate analysis of variance, estimates of genotypic variance, phenotypic variance, environmental variance, and covariances between pairs of traits, for a given number of sets of traits

4. Location of the program:

Program:

A:\CompBlom\94\Genocorr.exe

A:\CompBlom\94\Genocorr.for

(Software: FORTRAN Program)

5. Location of illustrative data and results:

Data:

A:\CompBlom\94\Genocorr.txt

Results:

A:\CompBlom\94\Genocorr.out

6. Client(s)

Plant breeders

7. Date

June 1994 (First edition January 1991)

8. Comments

M. Singh

Program:

```

    REAL HH(31)
    REAL ZZ(31,4,28), X(4,28),Y(4,28)
    INTEGER IN(4), IOUT(4)
    DATA IN /3,10,17,25/
    DATA IOUT /9,16,24,31/
C=====
C  FOR A STUDENT FROM A NARS
    TYPE*, ' NV,NB , NRN'
    READ(1,*) NV,NB,NRN
    NTOT=NV*NB
    DO 50 I=1,NTOT
    READ (5,*) I1,I2,(HH(K),K=3,31)
    DO 55 J=1,NB
    DO 55 K=1,NV
55  ZZ(K,I1,I2)=HH(K)
    50 CONTINUE
    DO 65 ISET=1,4
    WRITE (6,911) ISET
911  FORMAT(/,'***** SET OF VARIABLES**** = ',I5,/)
    IOUT0=IOUT(ISET)
    IOUT1=IOUT(ISET)-1
    IN0=IN(ISET)
    DO 60 K=IN0,IOUT1
    K1=K+1
    DO 60 KK=K1,IOUT0
    WRITE (6,912) K,KK
912  FORMAT(/,'***** NUMBERS OF VARIABLES IN PAIR = ',I4,I4,/)
    DO 70 J=1,NB
    DO 70 I=1,NV
    Y(J,I)=ZZ(K,J,I)
70  X(J,I) =ZZ(KK,J,I)
    CALL CORR(NV,NB,X,Y,NRN)
60  CONTINUE
65  CONTINUE
C 10 CONTINUE
    STOP
    END
    SUBROUTINE CORR( NV,NB,X,Y,NRN)
    REAL X(4,30),Y(4,30), TX(30),TY(30),BX(4),BY(4)
    V=NV
    B=NB
    DO 50 J=1,NB
    BX(J)=0.
50  BY(J)=0.
    DO 60 I=1,NV
    TX(I)=0.
60  TY(I)=0.
    GX=0.
    GY=0.
    GXX=0.

```

```

GYY=0.
GXY=0.
DO 100 J=1,NB
DO 100 I=1,NV
YIJ=Y(J,I)
XIJ=X(J,I)
BX(J)=BX(J)+XIJ
BY(J)=BY(J)+YIJ
TX(I)=TX(I)+XIJ
TY(I)=TY(I)+YIJ
GX=GX+XIJ
GY=GY+YIJ
GXX=GXX+XIJ*XIJ
GYY=GYY+YIJ*YIJ
GXY=GXY+XIJ*YIJ
100 CONTINUE
CFX=GX*GX/V/B
CFY=GY*GY/V/B
CFXY=GX*GY/V/B
GXX=GXX-CFX
GYY=GYY-CFY
GXY=GXY-CFXY
BSX=0.
BSY=0.
BSXY=0.
TSX=0.
TSY=0.
TSXY=0.
DO 110 J=1,NB
BSX=BSX+BX(J)*BX(J)
BSXY=BSXY+BX(J)*BY(J)
110 BSY=BSY+BY(J)*BY(J)
DO 120 I=1,NV
HX=TX(I)/B
HY=TY(I)/B
TSX=TSX+TX(I)*TX(I)
TSXY=TSXY+TX(I)*TY(I)
120 TSY=TSY+TY(I)*TY(I)
BSX=BSX/V-CFX
BSY=BSY/V-CFY
TSX=TSX/B-CFX
TSY=TSY/B-CFY
BSXY=BSXY/V-CFXY
TSXY=TSXY/B-CFXY
DFB=B-1
DFT=V-1
DFTO=V*B-1
DFE=(B-1)*(V-1)
ERRX=GXX-BSX-TSX
ERRY=GYY-BSY-TSY
ERRXY=GXY-BSXY-TSXY

```

```

SGEX=ERRX/DFE
SGEY=ERRY/DFE
SGEXY=ERRXY/DFE
SGGX=(TSX/DFT-SGEX)/B
SGGY=(TSY/DFT-SGEY)/B
SGGXY=(TSXY/DFT-SGEXY)/B
SGPX=SGGX+SGEX
SGPY=SGGY+SGEY
SGPXY=SGGXY+SGEXY
ROG=-99.
IF((SGGX.GT.0.).AND.(SGGY.GT.0.)) ROG=SGGXY/SQRT(SGGX*SGGY)
ROE=SGEXY/SQRT(SGEX*SGEY)
ROPA=TSXY/SQRT(TSX*TSY)
ROP=-99.
IF((SGPX.GT.0.).AND.(SGPY.GT.0.)) ROP=SGPXY/SQRT(SGPX*SGPY)
C   WRITE(7,800) NV,NB,SGGX,SGGY,SGEX,SGEY,ROG,ROE,ROP,ROPA
C 800 FORMAT(10X, 2(I4,X),3X,4(F12.5,X),4(F10.6,X))
      WRITE(6,900)
C   WRITE(6,*) DFB,BSX,BSY,BSXY
C   WRITE(6,*) DFT,TSX,TSY,TSXY
C   WRITE(6,*) DFE,SGEX,SGEY,ERRXY
      BSX=BSX/DFB
      BSY=BSY/DFB
      TSX=TSX/DFT
      TSY=TSY/DFT
900  FORMAT( X,' ***** BIVARIATE ANALYSIS OF VARIANCE ***')
      BSXY=BSXY/DFB
      TSXY=TSXY/DFT
      WRITE(6,*) DFB,BSX,BSY,BSXY
      WRITE(6,*) DFT,TSX,TSY,TSXY
      WRITE(6,*) DFE,SGEX,SGEY,SGEXY
      WRITE(6,*) SGPX,SGPY,SGPXY
      WRITE(6,*) NV,NB,SGGX,SGGY,SGEX,SGEY,ROG,ROE,ROP,ROPA
      HX2=SGGX/(SGGX+SGEX)
      HY2=SGGY/(SGGY+SGEY)
      WRITE( 6,*) HX2,HY2
      CALL SIMG(NV,NB,HX2,HY2,ROE,ROG,NRN)
      CALL SIMP(NV,NB,HX2,HY2,ROE,ROP,NRN)
      RETURN
      END
      SUBROUTINE SIMG(NV,NB,HX2,HY2,ROE,ROG,NRN)
      REAL ST(2),VR(2),RM(2,4)
      REAL PLEV(10),RLEV(2,10),HV1(5000),HV2(5000)
      DATA PLEV/ .005,.01,.025,.05,.1,.90,.95,.975,.99,.995/
C
C   *****
C   This presents simulated significance levels and
c   simulated moments of r(g), approximate variances, and
c   probability of invalid estimates of rho(g) when population
c   rho(g)=0.
c   Bivariate normal distribution assumed.

```

```

c .....
C
NPT=10
NST=2
ZMN=0.
ZSD=1.
CALL G05CBF(0)
B=NB
V=NV
Q=V-1.
F=Q*(B-1.)
NQ=Q
NF=F

C
c      note :  above vhx , vhy are in square form
c      vhx,vhy =var(g)/( var(g)+var(e) )
C      APPROXIMATION
XL2=1./HX2-1.
YL2=1./HY2-1.
XL=SQRT(XL2)
YL=SQRT(YL2)
HX=SQRT(HX2)
HY=SQRT(HY2)
SG1=SQRT(B+XL2)
SG2=SQRT(B+YL2)
SE1=XL
SE2=YL

C      approximations : untransformed, arcsin
C      Genotype correlation r(g) : VR(1) , VR(2)
ROE2=ROE*ROE
VR(1)=1./Q+(XL2+YL2)/B/Q+(1.+ROE2)*XL2*YL2/B/F
VR(2)=VR(1)
ROPA=ROE*XL*YL/(SG1*SG2)

C      _____ SIMULATION PART _____
C
C
DO 155 I=1,NST
DO 156 J=1,4
156 RM(I,J)=0.
155 CONTINUE
NVC=0
DO 555 II=1,NRN
C  GENERATE WISHART VARIABLES GXX, GYY, GXY WITH NQ, SG1, SG2, ROPA
N=NQ
RHO=ROPA
SGX=SG1
SGY=SG2

c
XX=0.
YY=0.
XY=0.

```

```

DO 100 I=1,N
H1=G05DDF(ZMN,ZSD)
H2=G05DDF(ZMN,ZSD)
H2=(RHO*H1+H2*SQRT(1.-RHO*RHO))*SGY
H1=H1*SGX
XX=XX+H1*H1
YY=YY+H2*H2
XY=XY+H1*H2
100 CONTINUE
GXX=XX
GYY=YY
GXY=XY
C  GENARATE WISHART VARIABLES EXX,EYY,EXY WITH NF,SE1,SE2,ROE
RHO=ROE
N=NF
SGX=SE1
SGY=SE2
C
XX=0.
YY=0.
XY=0.
DO 200 I=1,N
H1=G05DDF(ZMN,ZSD)
H2=G05DDF(ZMN,ZSD)
H2=(RHO*H1+H2*SQRT(1.-RHO*RHO))*SGY
H1=H1*SGX
XX=XX+H1*H1
YY=YY+H2*H2
XY=XY+H1*H2
200 CONTINUE
EXX=XX
EXY=XY
EYY=YY
C
D1=(GXX/Q-EXX/F)/B
D2=(GYY/Q-EYY/F)/B
IF((D1.LE.0.).OR.(D2.LE.0.)) GOTO 555
H=(GXY/Q-EXY/F)/B/SQRT(D1*D2)
C
IF(ABS(H).GT.1.) GOTO 555
ST(1)=H
ST(2)=ASIN(H)
NVC=NVC+1
HV1(NVC)=ST(1)
HV2(NVC)=ST(2)
C
DO 240 I=1,NST
HH=1.
STI=ST(I)
DO 250 J=1,4
HH=HH*STI

```

```

      RM(I,J)=RM(I,J)+HH
250  CONTINUE
240  CONTINUE
555  CONTINUE
      PERC=1.-FLOAT(NVC)/FLOAT(NRN)
C
      NNRN=NVC
      FNRN=NNRN
      IF(NVC.LE.1) GOTO 1
C
      DO 260 I=1,NST
      DO 270 J=1,4
270  RM(I,J)=RM(I,J)/FNRN
      RM(I,3)=RM(I,3)/RM(I,2)/SQRT(RM(I,2))
      RM(I,4)=RM(I,4)/RM(I,2)/RM(I,2)-3.
260  CONTINUE
C      write mean=bias as 10* mean ; SD= 10* standard deviation
      DO 280 I=1,NST
      RM(I,1)=10.*RM(I,1)
280  RM(I,2)=10.*SQRT(RM(I,2))
      VR(1)=10.*SQRT(VR(1))
290  VR(2)=10.*SQRT(VR(2))
C
C      ARRANGE HV1, HV2 IN ASCENDING ORDER OF MAGNITUDE
      CALL ORDER(NVC,HV1)
      CALL ORDER(NVC,HV2)
C
      DO 320 K=1,NPT
      INT1=FNRN*PLEV(K)
      RLEV(1,K)=HV1(INT1)
      RLEV(2,K)=HV2(INT1)
320  CONTINUE
      WRITE(6,906) HX2,HY2,ROE,NV,NB,PERC,RM(1,1),RM(1,2),VR(1),
+RM(1,3),RM(1,4),(RLEV(1,J),J=1,NPT)
      WRITE(6,907) RM(2,1),RM(2,2),VR(2),RM(2,3),RM(2,4),
+(RLEV(2,J),J=1,NPT)
906  FORMAT(2X,2(F3.1,X),F3.1,X,I3,X,I2,X,F4.2,
+3X,F5.2,X,F5.3,X,F5.2,X,F5.2,X,F5.2,3X,10(F5.3,X))
907  FORMAT(28X,F5.2,X,F5.2,X,F5.2,X,F5.2,X,F5.2,3X,10(F5.2,X))
1  RETURN
END
SUBROUTINE SIMP(NV,NB,HX2,HY2,ROE,ROP,NRN)
REAL ST(2),VR(2),RM(2,4)
REAL PLEV(10),RLEV(2,10),HV1(5000),HV2(5000)
DATA PLEV/ .005,.01,.025,.05,.1,.90,.95,.975,.99,.995/
C *****
C      This presents simulated moments ,critical values , approximate
C      variances of rho(pb), and probability of invalid estimates
C      of rho(pb) when population rho(pb) =0.
C      Bivariate normal distribution assumed .
C *****

```

```

C
NPT=10
NST=2
ZMN=0.
ZSD=1.
CALL G05CBF(0)
B=NB
V=NV
Q=V-1.
F=Q*(B-1.)
NQ=Q
NF=F

c      note : above vhx , vhy are in square form
c      vhx,vhy =var(g)/( var(g)+var(e) )
C      APPROXIMATION

XL2=1./HX2-1.
YL2=1./HY2-1.
XL=SQRT(XL2)
YL=SQRT(YL2)
HX=SQRT(HX2)
HY=SQRT(HY2)
SG1=SQRT(B+XL2)
SG2=SQRT(B+YL2)
SE1=XL
SE2=YL

C      approximations : untransformed, arcsin
C      Phenotype correlation r(gb) : VR(1),VR(2)
ROG=-ROE*XL*YL
VR(1)=(1.+(B-1.)*(1.+ROG*ROG)*HX2*HY2)/B/Q
VR(2)=VR(1)
ROPA=(1.-B)*ROE*XL*YL/SG1/SG2

C      _____ SIMULATION PART _____
C
C
DO 155 I=1,NST
DO 156 J=1,4
156 RM(I,J)=0.
155 CONTINUE
NVC=0
DO 555 II=1,NRN
C  GENERATE WISHART MATRIX GXX,GYY,GXY WITH NQ, SG1,SG2,ROPA
N=NQ
RHO=ROPA
SGX=SG1
SGY=SG2
XX=0.
YY=0.
XY=0.
DO 100 I=1,N
H1=G05DDF(ZMN,ZSD)
H2=G05DDF(ZMN,ZSD)

```

```

H2=(RHO*H1+H2*SQRT(1.-RHO*RHO))*SGY
H1=H1*SGX
XX=XX+H1*H1
YY=YY+H2*H2
XY=XY+H1*H2
100 CONTINUE
GXX=XX
GYY=YY
GXY=XY
C   GENERATE WISHART MATRIX EXX,EXY,EYY WITH NF, SE1,SE2,ROE
N=NF
RHO=ROE
SGX=SE1
SGY=SE2
XX=0.
YY=0.
XY=0.
DO 200 I=1,N
H1=G05DDF(ZMN,ZSD)
H2=G05DDF(ZMN,ZSD)
H2=(RHO*H1+H2*SQRT(1.-RHO*RHO))*SGY
H1=H1*SGX
XX=XX+H1*H1
YY=YY+H2*H2
XY=XY+H1*H2
200 CONTINUE
EXX=XX
EXY=XY
EYY=YY
C
D1=(GXX/Q-EXX/F)/B
D2=(GYY/Q-EYY/F)/B
IF((D1.LE.0.).OR.(D2.LE.0.)) GOTO 555
H=(GXY+EXY)/SQRT((GXX+EXX)*(GYY+EYY))
IF(ABS(H).GT.1.) GOTO 555
ST(1)=H
ST(2)=ASIN(H)
NVC=NVC+1
HV1(NVC)=ST(1)
HV2(NVC)=ST(2)
C
DO 240 I=1,NST
HH=1.
STI=ST(I)
DO 250 J=1,4
HH=HH*STI
RM(I,J)=RM(I,J)+HH
250 CONTINUE
240 CONTINUE
555 CONTINUE
PERC=1.-FLOAT(NVC)/FLOAT(NRN)

```

```

FNRN=NVC
IF(NVC.LE.1) GOTO 1
C
DO 260 I=1,NST
DO 270 J=1,4
270 RM(I,J)=RM(I,J)/FNRN
RM(I,3)=RM(I,3)/RM(I,2)/SQRT(RM(I,2))
RM(I,4)=RM(I,4)/RM(I,2)/RM(I,2)-3.
260 CONTINUE
c write mean=bias as 10* mean ; SD= 10* standard deviation
DO 280 I=1,NST
RM(I,1)=10.*RM(I,1)
280 RM(I,2)=10.*SQRT(RM(I,2))
VR(1)=10.*SQRT(VR(1))
VR(2)=10.*SQRT(VR(2))
C
CALL ORDER(NVC,HV1)
CALL ORDER(NVC,HV2)
DO 320 K=1,NPT
INT1=FNRN*PLEV(K)
RLEV(1,K)=HV1(INT1)
RLEV(2,K)=HV2(INT1)
320 CONTINUE
WRITE(6,908) HX2,HY2,ROE,NV,NB,PERC,RM(1,1),RM(1,2),VR(1),
+RM(1,3),RM(1,4),(RLEV(1,J),J=1,NPT)
WRITE(6,909) RM(2,1),RM(2,2),VR(2),RM(2,3),RM(2,4),
+(RLEV(2,J),J=1,NPT)
908 FORMAT(2X,2(F3.1,X),F3.1,X,I3,X,I2,X,F4.2,
+3X,F4.2,X,F5.3,X,F5.3,X,F5.2,X,F5.2,3X,10(F5.3,X))
909 FORMAT(28X,F4.2,X,F5.3,X,F5.3,X,F5.2,X,F5.2,3X,10(F5.2,X))
1 RETURN
END
SUBROUTINE ORDER(N,X)
REAL X(5000)
C ARRANGE X IN ASCENDING ORDER OF MAGNITUDE
DO 5 I=2,N
IF (X(I)-X(I-1))1,5,5
1 TEMP=X(I)
IM=I-1
DO 3 J=1,IM
L=I-J
IF (TEMP-X(L))2,4,4
2 X(L+1)=X(L)
3 CONTINUE
X(1)=TEMP
GOTO 5
4 X(L+1)=TEMP
5 CONTINUE
RETURN
END

```

Output Not included

**Computer and Biometrics Services Unit
ICARDA**

Serial Number -003

GENSTAT Modules in Computing Biometrics

Title: Analysis of data from Diallel experiments-I

1. Introduction:

This program analyses data from a diallel experiment using

- i) Hayman's method: to estimate components of variation, Vr-Wr graphs,
- ii) Griffing's method III and both models for estimating combining ability effects and genetic variance components.

2. Areas of application and keywords:

Plant breeding and genetics, Crop improvement.

3. Input/Output:

Input:

Plot-wise values of replications, blocks, first parent, second parent, variable(s).

Output:

Griffing's approach: Gca, sca's estimates and se's, estimates of gca and sca variances
Hayman's approach: Estimates of components of variation and se's, Parent -offsprings array variances/covariances (Vr -Wr), tests on validity of model, estimate of products of the two alleles frequencies, number of dominant genes, heritability in broad sense and narrow sense etc. These for F1's data and also for F2's data;

Heterosis and inbreeding coefficients.

4. Location of the program:

Program:

A:\CompBiom\94\Diallel1.gen
(Software: GENSTAT 5)

5. Location of illustrative data and results:

Data:

A:\CompBiom\94\diallel1.Txt

Results:

A:\CompBiom\94\diallel1.opt

6. Client(s)

Mr. M. Labdi and Dr. K. B. Singh

7. Date

February 1994

8. Comments

M. Singh

Program:

.....

Estimation of components of variation using
 - Hayman's approach and
 - Griffing's approach.

.....

.....

"

Scal NRep,NBlk,Np, Coeff ; 2,8,8,1

Scal NG, NGAll

Calc NG=Int(Np*(Np+1)/2)

Calc NGAll=Np*Np

Calc Unit=NGAll*NRep

Unit[Unit]

" Read Data from Labdi's file "

Fact[leve=NRep]Rep:&[leve=NBlk]Blk

Fact[leve=NGAll;labe=lt(P1,F12,F13,F14,F15,F16,F17,F18, \
 P2,F23,F24,F25,F26,F27,F28, P3,F34,F35,F36,F37,F38, \
 P4,F45,F46,F47,F48, P5,F56,F57,F58, P6,F67,F68, P7,F78, P8, \
 S12,S13,S14,S15,S16,S17,S18,\
 S23,S24,S25,S26,S27,S28, S34,S35,S36,S37,S38, \
 S45,S46,S47,S48, S56,S57,S58, S67,S68, S78)] Geno

Open 'Diallel1.txt' ; ch=2; fi=in

Scal Nlines

Read[Ch=2]Nlines

Skip[ch=2]Nlines

Read[ch=2] Rep,Blk,Geno,NPlants, SC1,SC1%, SC2,SC2%, SC3,SC3% ;frep=le,le,le

Scal Sigma2

Scal SigRep,SigBlk,DfRep,DfBlk,DfError

Scalar E

Scal DfPlotEr

Vari[nval=NGAll] GMean0

```
Var[nval=NG]GMean
```

```
For Yield = SC1 ",SC1%, SC2,SC2%, SC3,SC3% "
```

```
Vcomp[fixed=Rep+Geno]Rep/Blk+Geno
```

```
Reml Yield
```

```
Vkeep[sigma2=Sigma2] Geno; Means=MT
```

```
Equa MT; GMean0
```

```
Calc E=Sigma2/NRep
```

```
Calc DfRep=NRep-1: & DfBlk=NRep*(NBlk-1)
```

```
Calc DfPlotEr=NRep*NGAll-DfRep-DfBlk-(NGAll-1)
```

```
Var[Nvalu=NG]ParF1,ParF2,GMean
```

```
Scal ij,ij1
```

```
Calc ij=1
```

```
For i=1...Np
```

```
For j=1...Np
```

```
Calc ParF1$(ij)=GMean0$(ij)
```

```
If i.eq.j
```

```
Calc ParF2$(ij)=ParF1$(ij)
```

```
else
```

```
Calc ij1=Np*(Np+1)/2+ij-i
```

```
Calc ParF2$(ij)=GMean0$(ij1)
```

```
endif
```

```
Calc ij=ij+1
```

```
Endf:Endf
```

```
" ***** Generate Parent levels ***** "
```

```
Fact[leve=Np;Nval=NG]P1,P2
```

```
Scal ij
```

```
Calc ij=0
```

```
For i=1...Np : For j=1...Np
```

```
Calc ij=ij+1
```

```
Calc P1$(ij)=Int(i) : & P2$(ij)=Int(j)
```

```
Endf : Endf
```

```
Print P1,P2,ParF1,ParF2
```

```
"***** Griffing's method ***** "
```

```
Var [nval=NG]G[1...Np],SCAEff[1,2]
```

```
For i=1...Np;GG=G[1...Np]
```

```
Calc GG=(P1.eq.i)+(P2.eq.i)
```

```
Endf
```

```

Scal Mn
For Dum=ParF1,ParF2; SCA=SCAEff[1,2]
Calc Mn=Mean(Dum)

```

" Griffing model I "

```

Calc Dum=Dum-Mn
Model Dum ; fitt=SCA
Fit [prin=m,s;fpro=y] G[1...Np]
Rkeep DF=df;devi=ss
Fit [cons=o;prin=m,e] G[1...Np]
Calc SCA=Dum-SCA

```

" Griffing model II "

```

Scal SeS2G,SeS2S,SeS2A,SeS2D,Sigma2G,Sigma2S,Sigma2A,Sigma2D,GCAMS
Calc Sigma2S=ss/df-Sigma2/NRep
Calc GCAMS=((Nobs(Dum)-1)*Var(Dum)-ss)/(Np-1): & Sigma2G=(GCAMS-ss/df)/(Np+2)
Calc SeS2G=sqrt( (2*GCAMS**2+4*(ss/df)**2/Np/(Np+2))/(Np-1)/(Np+2) )
Calc SeS2S=sqrt( 4*(ss/df)**2/Np/(Np-1)+2*(Sigma2/NRep)**2/DfPlotEr )
Calc Sigma2A=2*Sigma2G
Calc SeS2A=2*SeS2G
Calc Sigma2D=Sigma2S
Calc SeS2D=SeS2S
Print E,Sigma2G,SeS2G,Sigma2S,SeS2S,Sigma2A,SeS2A,Sigma2D,SeS2D
Calc Dum=Dum+Mn
Endf
Rest ParF1,SCAEff[1],ParF2,SCAEff[2] ; cond=P1.ne.P2
print P1,P2,ParF1,SCAEff[1],ParF2,SCAEff[2] ;file=10
Rest P1,P2,ParF1,SCAEff[1],ParF2,SCAEff[2]
Scal SeGca, SeSca
Calc SeGca=sqrt((Np-1)*Sigma2/NRep/Np/(Np+2))
Calc SeSca=sqrt((Np*Np+Np+2)*Sigma2/NRep/(Np+1)/(Np+2)) "lines i <> j unequal"
Prin SeGca, SeSca

```

" Parents + F1 s "

```
Calc GMean=ParF1
```

"

***** Notations used *****

1. Statistics available

P1, P2, GMean : vectors (NG) of first parent, second parent and
resulting offspring (adjusted) mean
(over blocks/replicates) respectively

VoLo Variance of parent means (parent variability)

Vr Vector (Np) of variance of each of Np arrays

V1L1 Mean(Vr): mean of array variances (bar Vr)

Wr Vector (Np) of covariances between parents and

WoLo1 their offsprings for each array
 Mean(Wr): mean of Np covariances in Wr
 VoL1 Variance of means of arrays
 ML1 mean of parents
 MLo mean of their nxn progenies

2. Variance components to be estimated

E sigma2/rep
 expected environmental component of variation

D = VoLo-E
 component of variation due to additive effects

H1 =VoLo -4WoLo1+4V1L1 -(3Np-2)E/(n

MDD: degree of dominance

GenesPN: uv average: Positive negative gene freq product

GenesDR : proportion of dominant and recessive genes

CorrP : correlation between parental order of dominance
and parental measurements

NGGD: number of groups of genes exhibiting dominance

h2ns : heritability in narrow sense

h2bs : heritability in broad sense

"

" *****Variance of parents VoLo ***** "

Scalar VoLo

Rest GMean; Cond=P1.eq.P2

Calc VoLo=Var(GMean)

Rest GMean

***** Compute Vr , Wr, VrWr *****

Vari[Nval=Np]Vr,Wr,WrVr,Fr,ArM,Vhelp,PMean,WrLim,WrPred,Wrb1

Fact[leve=Np;valu=1...Np; nval=Np]Parents

For i=1...Np

Rest GMean; Cond=P1.eq.i.or.P2.eq.i

Calc Vr[i]=Var(GMean)

Calc ArM[i]=Mean(GMean)

Rest GMean

Rest GMean; Cond=P1.eq.i.and.P2.eq.i; saveset=set

Calc PMean[i]=GMean[set]

Rest GMean

Endf

" Compute V1L1, VoL1,WoLo1 "

Scal V1L1,VOL1,WoLo1

Calc V1L1=Mean(Vr)

Calc VoL1=Var(ArM)

For i=1...Np

Rest GMean; Cond=P1.eq.i.or.P2.eq.i; saveset=set

Calc Vhelp=GMean\$[set]

Calc Wr\$[i]=Covariance(PMean;Vhelp)

Endf

Calc WoLo1=Mean(Wr)

Scal ML1,MLO

Rest GMean; Cond=P1.eq.P2

Calc ML1=Mean(GMean)

Rest GMean; Cond=P1.ne.P2

Calc MLo=Mean(GMean)

Calc MLo=((Np-1)*MLO+ML1)/Np

Rest GMean

Calc WrVr=Wr+Vr

Calc WrLim=Sqrt(Vr*VoLo)

Model Wr; Fitt=WrPred

Fit Vr

Rkeep Est=Est; Vcov=Vcov;DF=DF

Scal Int, Slope

Equat Est; Ip(Int,Slope)

Calc Fr=2*(VoLo-WoLo1+V1L1-Wr-Vr)-2*(Np-2)*E/Np

Scal Tsq, Prob

" 1. Test b=1 "

Calc Tsq =(Slope-1)**2/Vcov\$[2;2]

Calc Prob=1-Fpro(Tsq;1;DF)

Print Tsq,DF,Prob

" 2. Test the Validity of Hypothesis"

Calc Tsq=(Np-2)*(Var(Vr)-Var(Wr))**2/(Var(Vr)*Var(Wr)- \

(Covariance(Wr;Vr))**2)/4

Calc Prob=1-Fpro(Tsq;4;DF)

Prin Tsq,DF,Prob

" Components of variation"

```

Scal D,F,H1,H2,h2,s2,Np2,Np3
Calc Np2=Np*Np : & Np3=Np*Np2
Calc D=VoLo-E :& F=2*VoLo-4*WoLo1-2*(Np-2)*E/Np
Calc H1=VoLo-4*WoLo1+4*V1L1-(1+4*(Np-1)/Np/Coeff)*E
Calc H2=4*V1L1-4*VoL1-4*(Np-1)*(1+(Np-1)/Coeff)*E/Np2
Calc h2=4*(ML1-MLo)**2-4*(Np-1)*E/Np2

***** Standard errors *****

Calc s2=0.5*Var((Wr-Vr))
Calc SeD=Sqrt(s2*(1+1/Np)) :& SeF=Sqrt(s2*(4+20/Np-16/Np2+16/Np3))
Calc SeH1=Sqrt(s2*(1+41/Np-12/Np2+4/Np3)) : & SeH2=sqrt(s2*(36/Np))
Calc Seh2=Sqrt(s2*(16+16/Np2-32/Np3+16/Np/Np3)/Np)
Calc SeE=sqrt(s2/Np)
Print s2, D,SeD,F,SeF,H1,SeH1,H2,SeH2,h2,Seh2,E,SeE

Calc WrB1=D/4-H1/4+Vr
Print Parents,Vr,Wr,WrVr,Fr,ArM,PMean,WrLim,WrPred,WrB1
Print VoLo,V1L1,VoL1,WoLo1,ML1,MLo
Graph WrB1,WrLim,WrPred,Wr,Vr; symb=Parents; meth=c,c,c,p
Graph WrVr,PMean; symb=Parents; meth=p

" Other parameters on genes "
Scal MDD,GenesPN,GenesDR,CorrP, NGGD,h2ns,h2bs

Calc MDD=sqrt(H1/D) : & GenesPN=H2/4/H1
Calc GenesDR=(sqrt(4*D*H1)+F)/(sqrt(4*D*H1)-F)
Calc NGGD=h2/H2
Calc CorrP= Correlation(WrVr,PMean)

" Recalculate F as mean of Fr"
Calc F=Mean(Fr)

Calc h2ns=(0.5*D+0.5*H1-.5*H2-.5*F)/(.5*D+.5*H1-.25*H2-.5*F+E)
Calc h2bs=(0.5*D+0.5*H1-.25*H2-.5*F)/(.5*D+.5*H1-.25*H2-.5*F+E)

Prin MDD,GenesDR,GenesPN,CorrP, NGGD,h2ns,h2bs, F

" Most dominant and recessive parents"

Scal XPlus, XMinus, AFreq,aFreq, YDom,YRec
Scal YD,YR
Calc XPlus=(1+sqrt(1-4*(WoLo1-V1L1)/VoLo))/2
Calc XMinus=(1-sqrt(1-4*(WoLo1-V1L1)/VoLo))/2

print XPlus,XMinus

If XPlus.gt.1
  Calc XPlus=XPlus-1
Else

```

```
Endif

If XMinus.gt.1
    Calc XMinus=XMinus-1
Else
Endif

If XPlus.lt.0
    Calc XPlus=0.
Else
Endif

If XMinus.lt.0
    Calc XMinus=0.
Else
Endif

print XPlus,XMinus

Scal XPlus0,XMinus0
Calc XPlus0=XPlus: Calc XMinus0=1-XPlus0
Calc YDom=ML1+Slope*(VoLo*XMinus0*(XMinus0+1)-WoLo1-V1L1)
Calc YRec=ML1+Slope*(VoLo*XPlus0*(XPlus0+1)-WoLo1-V1L1)
Print XPlus0, XMinus0,YDom,YRec

Calc XMinus0=XMinus :Calc XPlus0=1-XMinus0
Calc YDom=ML1+Slope*(VoLo*XMinus0*(XMinus0+1)-WoLo1-V1L1)
Calc YRec=ML1+Slope*(VoLo*XPlus0*(XPlus0+1)-WoLo1-V1L1)
Print XPlus0, XMinus0,YDom,YRec

" Parents + F2 's ***** "
```

Calc GMean=ParF2

" *****Variance of parents VoLo ***** "

```
Scalar VoLo
Rest GMean; Cond=P1.eq.P2
Calc VoLo=Var(GMean)
Rest GMean

***** Compute Vr , Wr, VrWr *****

Var[Nval=Np]Vr,Wr,WvVr,Fr,ArM,Vhelp,PMean,WvLim,WvPred
Fact[leve=Np;valu=1...Np; nval=Np]Parents
```

```

For i=1...Np
Rest GMean; Cond=P1.eq.i.or.P2.eq.i
Calc Vr[i]=Var(GMean)
Calc ArM[i]=Mean(GMean)
Rest GMean
Rest GMean; Cond=P1.eq.i.and.P2.eq.i; saveset=set
Calc PMean[i]=GMean[set]
Rest GMean
Endf

" ***** Compute V1L1, VoL1,WoLo1 ***** "

Scal V1L2,VoL2,WoLo2
Calc V1L2=Mean(Vr)
Calc VoL2=Var(ArM)

For i=1...Np
Rest GMean; Cond=P1.eq.i.or.P2.eq.i; saveset=set
Calc Vhelp=GMean[set]
Calc Wr[i]=Covariance(PMean;Vhelp)
Endf
Calc WoLo2=Mean(Wr)

Scal ML1,MLo

Rest GMean; Cond=P1.eq.P2
Calc ML1=Mean(GMean)
Rest GMean; Cond=P1.ne.P2
Calc MLo=Mean(GMean)
Calc MLo=((Np-1)*MLo+ML1)/Np
Rest GMean

Calc WrVr=Wr+Vr
Calc WrLim=Sqrt(Vr*VoLo)
Model Wr; Fitt=WrPred
Fit Vr
Rkeep Est=Est; Vcov=Vcov;DF=DF
Scal Int, Slope
Equat Est; Ip(Int,Slope)

Print VoLo,V1L2,VoL2,WoLo2,ML1,MLo
Scal Tsq, Prob

" 1. Test b=1 "
Calc Tsq =(Slope-1)**2/Vcov$[2;2]
Calc Prob=1-Fpro(Tsq;1;DF)
Print Tsq,DF,Prob

" 2. Test the Validity of Hypothesis"

Calc Tsq=(Np-2)*(Var(Vr)-Var(Wr))**2/(Var(Vr)*Var(Wr)- \

```

```

      (Covariance(Wr;Vr))**2)/4
Calc Prob=1-Fpro(Tsq;4;DF)
Prin Tsq,DF,Prob

```

" ***** Components of variation *****"

```

Scal D,F,H1,H2,h2,s2,Np2,Np3
Calc Np2=Np*Np : & Np3=Np*Np2
Calc D=VoLo-E :& F=4*VoLo-8*WoLo1-4*(Np-2)*E/Np
Calc H1=4*VoLo-16*WoLo2+16*V1L2-4*(5*Np-4)*E/Np
Calc H2=16*V1L2-16*VoL2-16*(Np-1)*E/Np
Calc h2=16*(ML1-MLo)**2-16*(Np-1)*E/Np

```

"***** Standard errors *****"

```

Calc s2=0.5*Var((Wr-Vr))
Calc SeD=Sqrt(s2*(1+1/Np)) :& SeF=Sqrt(s2*(16+80/Np-64/Np2+6/Np3))
Calc SeH1=Sqrt(s2*(16+656/Np-192/Np2+64/Np3)) : & SeH2=sqrt(s2*(576/Np))
Calc Seh2=Sqrt(s2*(256+256/Np2-512/Np3+256/Np/Np3)/Np)
Calc SeE=Sqrt(s2/Np)
Print s2,D,SeD,F,SeF,H1,SeH1,H2,SeH2,h2,Seh2,E,SeE

```

```

Calc WrB1=D/4-H1/4+Vr
Print Parents,Vr,Wr,WrVr,ArM,PMean,WrLim,WrPred,WrB1
Print VoLo,V1L1,VoL1,WoLo1,ML1,MLo
Graph WrB1,WrLim,WrPred,Wr;Vr; symb=Parents; meth=c,c,c,p
Graph WrVr;PMean; symb=Parents; meth=p

```

" Other parameters on genes "

```

Scal MDD,GenesPN,GenesDR,CorrP, NGGD,h2ns,h2bs

```

```

Calc MDD=sqrt(H1/4/D) : & GenesPN=H2/4/H1
Calc GenesDR=(sqrt(4*D*H1)/4+F/2)/(sqrt(4*D*H1)/4-F/2)
Calc NGGD=h2/H2
Calc CorrP= Correlation(WrVr;PMean)
Calc h2ns=D/4/(D/4+H1/16-F/8+E)

```

```

Prin MDD,GenesDR,GenesPN,CorrP, NGGD,h2ns

```

" Most dominant and recessive parents"

```

Scal XPlus, XMinus, AFreq,aFreq, YDom,YRec
Scal YD,YR
Calc XPlus= (1+sqrt(1-4*GenesPN))/2
Calc XMinus=(1-sqrt(1-4*GenesPN))/2

```

```

print XPlus,XMinus

```

```

If XPlus.gt.1
  Calc XPlus=XPlus-1
Else
Endif

If XMinus.gt.1
  Calc XMinus=XMinus-1
Else
Endif

If XPlus.lt.0
  Calc XPlus=0.
Else
Endif

If XMinus.lt.0
  Calc XMinus=0.
Else
Endif

print XPlus,XMinus

Scal XPlus0,XMinus0
Calc XPlus0=XPlus: Calc XMinus0=1-XPlus0
Calc YDom=ML1+Slope*(VoLo*XMinus0*(XMinus0+1)-WoLo2-V1L2)
Calc YRec=ML1+Slope*(VoLo*XPlus0*(XPlus0+1)-WoLo2-V1L2)

Print XPlus0, XMinus0,YDom,YRec

Calc XMinus0=XMinus :Calc XPlus0=1-XMinus0
Calc YDom=ML1+Slope*(VoLo*XMinus0*(XMinus0+1)-WoLo2-V1L2)
Calc YRec=ML1+Slope*(VoLo*XPlus0*(XPlus0+1)-WoLo2-V1L2)
Print XPlus0, XMinus0,YDom,YRec

***** Heterosis and inbreeding depression *****

Scal i1, Np1,HMPar,HMPar%,HBPar,HBPar%,IDep,IDep%, PrHMPar,PrHBPar,PrIDep
Calc Np1=Np-1
For i=1...Np1
  Calc i1=i+1
  For j=i1...Np
    Rest ParF1,ParF2; P1.eq.i.and.P2.eq.j
    Calc MidPar=(PMean$(i)+PMean$(j))/2
    Calc HMPar=Mean(ParF1)-MidPar
    Calc Fval=HMPar**2/(3.5*Sigma2/NRep)
    Calc PrHMPar=1-FPro(Fval;1;d)

  Calc HMPar%=(HMPar/MidPar)*100.

```

```

Calc SupPar=PMean$[i]*(PMean$[j].lt.PMean$[i])+ \
                PMean$[j]*(PMean$[j].ge.PMean$[i])
Calc HBPar=Mean(ParF1)-SupPar
Calc Fval=HBPar**2/(2.*Sigma2/NRep)
Calc PrHBPar=1-FPro(Fval;1;df)
Calc HBPar%=(HBPar/SupPar)*100.

Calc IDep=Mean(ParF1)-Mean(ParF2)
Calc Fval=IDep**2/(2.0*Sigma2/NRep)
Calc PrIDep=1-FPro(Fval;1;df)
Calc IDep%=(IDep/Mean(ParF1))*100.

Print i,j,HMPar,HMPar%,PrHMPar,HBPar,HBPar%,PrHBPar,IDep,IDep%,PrIDep; \
      flen=10;deci=3
Endf
Endf

Rest ParF1,ParF2

Endf

Clos
stop

```

Output: Not included here

**Computer and Biometrics Services Unit
ICARDA**

Serial Number -004

GENSTAT Modules in Computing Biometrics

Title: Comparing parallelism of non-linear curves-I

1. Introduction:

This program shows steps for comparing a number of (ten standard) non-linear curves over the levels of a grouping factor, when the model errors are homoscedastic. It provides the sum of squares due to common parameters, changes due to constant terms, changes due to slopes and change due to separate non-linear parameters. It has illustrations using data from experiments conducted by Pasture, Forage and Live-stock program.

2. Areas of application and keywords:

Comparing non-linear curves.

3. Input/Output:

Input:

Levels of grouping factor, values of x-variables, values of y-variable (variable to be modelles)

Output:

Model, summary of regression analysis, accumulated analysis, estimates and standard errors

4. Location of the program:

Program:

Diskette : \CompBiom\94\CurveHom.gen
(Software: GENSTAT 5)

5. Location of illustrative data and results:

Data:

Diskette : \CompBiom\94\CurveHom.Txt

Results:

Diskette : \CompBiom\94\CurveHom.opt

6. Client(s)

Dr. S. Christiansen,PFLP

7. Date

Mar 1994

8. Comments

M. Singh

Program

" ANALYSIS OF PART II. Germination test. repeated over 6 months: monthly and bi-weekly measurements on percentage of hard seeds above ground and percentage of hard seeds below ground.

AHARD : % hard seeds above ground

BHARD : % hard seeds below ground

These are means over reps and management.

Source: Dr. S. Christiansen

UNIT[84]

Open ' Scott_curve.txt'; ch=2; fi=in

FACT[LEVE=7; LABE=IT(V2614,V2647,V2650,V2660,LC,MR,MN)] ACC

READ[ch=2] ACC,TIME, AHARD,BHARD; FREP=LEV

VARI[VALU=(0,1,1.5,2,2.5,3.,3.5,4.,4.5,5.,5.5,6.)7] MONTH

MODEL AHARD ; fitt=f

TERMS MONTH*ACC

FITC[CURVE=QDQ;prin=m,s;fpro=y] MONTH

ADD[prin=m,s;fpro=y] ACC

ADD[prin=m,s;fpro=y] MONTH.ACC

ADD[PRIN=M,E,S,A; fpro=y;tpro=y;NONL=SEPARATE]

Vari[nvalu=12]Fit[1...7],Mon[1...7]

Equa lp(f,MONTH); lp(Fit[1...7],Mon[1...7])

Graph Fit[1...7];Mon[1...7] ; meth=c

Graph Fit[1...7];Mon[1...7] ; meth=c; symb='1','2','3','4','5','6','7'

" Model BHARD using logistic.

Since BHARD is not meaningful for MR,MN , so use the data for only five accessions.

Vari[nvalu=60] bhard,Month

Fact[leve=5;labe=lt(V2614,V2647,V2650,V2660,LC);Nvalu=60] Acc

Equat BHARD; bhard

Equat MONTH; Month

Equat ACC; Acc

MODEL bhard; fitt=fit

TERMS Acc*Month

FITC[CURVE=logistic;prin=m,s;fpro=y] Month

ADD[prin=m,s;fpro=y] Acc

ADD[prin=m,s;fpro=y] Month.Acc

ADD[PRIN=M,E,S,A; fpro=y;tpro=y;NONL=SEPARATE]

```
Equa lp(fit,Month); lp(Fit[1...5],Mon[1...5])  
Graph Fit[1...5];Mon[1...5] ; meth=c  
Graph Fit[1...5];Mon[1...5] ; meth=c; symb='1','2','3','4','5'
```

```
clos  
stop
```

Output : Not included here

**Computer and Biometrics Services Unit
ICARDA**

Serial Number -005

GENSTAT Modules in Computing Biometrics

Title: Comparing parallelism of non-linear curves-II

1. Introduction:

This program shows steps for comparing a number of (ten standard) non-linear curves over the levels of a grouping factor, when the model errors are heteroscedastic over the levels of grouping factor. It provides the sum of squares due to common parameters, changes due to constant terms, changes due to slopes and change due to separate non-linear parameters. It has illustrations using data from experiments conducted by Pasture, Forage and Live-stock program.

2. Areas of application and keywords:

Comparing non-linear curves.

3. Input/Output:

Input:

Levels of grouping factor, values of x-variables, values of y-variable (variable to be modelles)

Output:

Estimation and test for heterogeneous errors over levels of the grouping factor, model, summary of regression analysis, accumulated analysis, estimates and standard errors

4. Location of the program:

Program:

Diskett : \CompBiom\94\CurveHet.gen
(Software: GENSTAT 5)

5. Location of illustrative data and results:

Data:

Diskette : \CompBiom\94\CurveHet.Txt

Results:

Diskette : \CompBiom\94\CurveHet.opt

6. Client(s)

Dr. S. Christiansen, PFLP

7. Date

Mar 1994

8. Comments

M. Singh

Program:

* ANALYSIS OF PART II. Germination test. repeated over 6 months: monthly and bi-weekly measurements on percentage of hard seeds above ground and percentage of hard seeds below ground.
 AHARD : % hard seeds above ground
 BHARD : % hard seeds below ground
 These are means over reps and management.

Source: Dr. S. Christiansen

Error variances may be heterogeneous

*

UNIT[84]

Open ' Scott_curve.txt'; ch=2; fl=in

FACT[LEVE=7; LABE=IT(V2614,V2647,V2650,V2660,LC,MR,MN)] ACC

READ[ch=2] ACC,TIME, AHARD,BHARD; FREP=LEV

VARI[VALU=(0,1,1.5,2,2.5,3.,3.5,4.,4.5,5.,5.5,6.)7] MONTH

" Estimate error variances and test their homogeneity"

Scal NL ; 7

Scal h1,h2,h3,h4;0,0,0,0

Scal chisq, prob

For i=1...NL

Rest AHARD, Wet; ACC.eq.i

MODEL AHARD ; fitt=f

FITC[CURVE=QDQ;prin=m,s,e;fpro=y] MONTH

Rkeep devi=ss; df=df

Calc h1=h1+df*log(ss/df)

Calc h2=h2+ss

Calc h3=h3+df

Calc Wet=df/ss

Calc h4=h4+1/df

Rest AHARD, Wet

Endf

Calc chisq=(h3*log(h2/h3)-h1)/(1+(h4-1/h3)/3/(NL-1))

Calc prob=1-chisq(chisq,NL-1)

Print chisq,prob

MODEL [weight=Wet]AHARD ; fitt=f

TERMS MONTH*ACC

FITC[CURVE=QDQ;prin=m,s;fpro=y] MONTH

ADD[prin=m,s;fpro=y] ACC

ADD[prin=m,s;fpro=y] MONTH.ACC

ADD[PRIN=M,s,a,e; fpro=y;tpro=y;NONL=SEPARATE]

```
Varl[nvalu=12]Fit[1...7],Mon[1...7]  
Equa lp(f,MONTH); lp(Fit[1...7],Mon[1...7])  
Graph Fit[1...7];Mon[1...7] ; meth=c  
Graph Fit[1...7];Mon[1...7] ; meth=c; symb='1','2','3','4','5','6','7'
```

```
clos  
stop
```

Output: Not included

**Computer and Biometrics Services Unit
ICARDA**

Serial Number -006

GENSTAT Modules in Computing Biometrics

Title: **Analysis of long-term rotational trials: 1. Estimation of main effects and interactions.**

1. Introduction:

This program performs analysis of data from two-course rotation systems in terms of evaluating the main effects and interactions of the treatment factors and crop rotations.

2. Areas of application and keywords:

Long-term rotation trials, repeated measures, two-course rotations, barley rotations

3. Input/Output:

Input:

Factor values of treatment, crop rotation, year, variable(s)

Output:

Analysis of variance table, tables of means and their standard errors.

4. Location of the program:

Program:

Diskett : \CompBiom\94\LTRT1*.gen (12 files)

(Software: GENSTAT 5)

5. Location of illustrative data and results:

Data:

Diskette : \CompBiom\94\LTRT1*.txt (3 files)

Results:

Diskette : \CompBiom\94\LTRT1*.OUT (12 files)

6. Client(s)

Dr. M. Jones, FRMP

7. Date

1992/1993/1994

8. Comments

M. Singh

Programs:

File LTRT11.GEN

```
UNIT[ 288]
OPEN 'B_LVF1.TXT'; CH=2;FI=IN
FACT[ LEVE=2; Labels=It(TelHadya, Breda)] LOC
Fact[ leve=2; Labels=It(Odd, Even)] SERIES
FACT[LEVE=22] TREAT
FACT[ LEVE=3] REP
FACT[LEVE=6; Labe=It(Y83_84,Y84_85,Y85_86,Y86_87,Y87_88,Y88_89)] YEAR
READ [ch=2] TREAT, LOC,REP,YEAR,SERIES,GY,ST; frep=lev
```

" This prog is prepared for training on Long-term rotational trials,
7-18 Mar 93 conducted by FRMP, ICARDA.

File B_LVF1.TXT has data on 3 rotations (B/V,B/F,B/B0; Three fertility
treatments (N20:P60 0: 0,0:0 0:0,N20:P30 N20:P30 N20:30,0:0 N20:30);
two sites (TelHadya, Breda),
6 years (Y83_84,Y84_85...Y88_89). We generate 2 Series (Odd, Even), 3 cycles
(First, Second, Third).

```
"
FACT[LEVE=6; labe=It(B_P, B_PB, B_V, B_VB, B_F, B_B)] ROTA
CALC ROTA=NEWL(TREAT;I((1,2,3,4)4,5,5,6,6,6))
FACT[LEVE=3; Labe=It(First, Second, Third)] CYCLE
CALC CYCLE=NEWL(YEAR;I(1,1,2,2,3,3))
FACT[LEVE=3; Labe=It('-F','+F',F_Others)] FERT1
CALC FERT1=NEWL(TREAT;I(3,3,2,3, 3,3,3,3, 3,3,3,3, 3,3,1,3, 1,2, 3,2,1,2))
dele TREAT
PRINT LOC,REP, ROTA, YEAR, FERT1,GY,ST; fiel=5(13),2(6); deci=2

for i=1,2 : for j=3,5,6
REST GY,ST; cond=LOC.eq.i.AND.ROTA.EQ.j
Tabu[class= REP, FERT1, YEAR,SERIES] GY,ST; TOTA=TBGY,TBSY
print[ miss=' ' ] TBGY,TBSY; field=6; deci=2
endf: endf
```

clos: stop

File LTRT12.GEN

"

This prog is prepared for training on Long-term rotational trials,
7-18 Mar 93 conducted by FRMP, ICARDA.

File B_LVF1.TXT has data on
3 rotations (B/V,B/F,B/B0;

3 fertility treatments

(N20:P60 0: 0,0:0 0:0,N20:P30 N20:P30 N20:30,0:0 N20:30);

2 sites (TelHadya, Breda),

6 years (Y83_84,Y84_85...Y88_89).

We generate 2 Series (Odd, Even), 3 cycles (First, Second, Third).

**** Analysis of data for individual locations,years, rotations*****

```

UNIT[ 288]
OPEN 'B_LVF1.TXT'; CH=2;FI=IN
FACT[ LEVE=2; Labels=It(TelHadya, Breda)] LOC
Fact[ leve=2; Labels=It(Odd, Even)] SERIES
FACT[LEVE=22] TREAT
FACT[ LEVE=3] REP
FACT[LEVE=6; Labe=It(Y83_84,Y84_85,Y85_86,Y86_87,Y87_88,Y88_89)] YEAR
READ [ch=2] TREAT, LOC,REP,YEAR,SERIES,GY,ST; frep=lev

FACT[LEVE=6; labe=It(B_P, B_PB, B_V, B_VB, B_F, B_B)] ROTA
CALC ROTA=NEWL(TREAT;I((1,2,3,4)4,5,5,6,6,6,6))
FACT[LEVE=3; Labe=It(First, Second, Third)] CYCLE
CALC CYCLE=NEWL(YEAR;I(1,1,2,2,3,3))
FACT[LEVE=3; Labe=It('-F','+F',F_Others)] FERT1
CALC FERT1=NEWL(TREAT;I(3,3,2,3, 3,3,3,3, 3,3,3,3, 3,3,1,3, 1,2, 3,3,1,2))
dele TREAT

```

" FOR B/V,B/F and b/b"

```

for i=1: for j=1: for k=3
rest GY,ST;cond=LOC.eq.i.and.YEAR.eq.j.and.ROTA.eq.k

```

```

BLOC REP/FERT1
TREA FERT1
ANOV[SE=M;FPRO=Y; FACT=4] GY ; FITT=F1; RESI=R1
GRAPH R1;F1
endf: endf: endf

```

clos:stop

File LTRT13.GEN

" *****

This prog is prepared for training on Long-term rotational trials,
7-18 Mar 93 conducted by FRMP, ICARDA.

File B_LVF1.TXT has data on
3 rotations (B/V,B/F,B/B);
3 fertility treatments

(N20:P60 0: 0,0:0 0:0,N20:P30 N20:P30 N20:30,0:0 N20:30);
 2 sites (TelHadya, Breda),
 6 years (Y83_84,Y84_85...Y88_89).
 We generate 2 Series (Odd, Even), 3 cycles (First, Second, Third).

**** Analysis of data for individual locations, rotations*****

```

UNIT[ 288]
OPEN 'B_LVF1.TXT'; CH=2;FI=IN
FACT[ LEVE=2; Labels=It(TelHadya, Breda)] LOC
Fact[ leve=2; Labels=It(Odd, Even)] SERIES
FACT[LEVE=22] TREAT
FACT[ LEVE=3] REP
FACT[LEVE=6; Labe=It(Y83_84,Y84_85,Y85_86,Y86_87,Y87_88,Y88_89)] YEAR
READ [ch=2] TREAT, LOC,REP,YEAR,SERIES,GY,ST; frep=lev

```

```

FACT[LEVE=6; labe=It(B_P, B_PB, B_V, B_VB, B_F, B_B)] ROTA
CALC ROTA=NEWL(TREAT;l((1,2,3,4)4,5,5,6,6,6))
FACT[LEVE=3; Labe=It(First, Second, Third)] CYCLE
CALC CYCLE=NEWL(YEAR;l(1,1,2,2,3,3))
FACT[LEVE=3; Labe=It('-F','+F',F_Others)] FERT1
CALC FERT1=NEWL(TREAT;l(3,3,2,3, 3,3,3,3, 3,3,3,3, 3,3,1,3, 1,2, 3,3,1,2))
dele TREAT

```

" FOR B/V,B/F and b/b"

```

for i=1: for k=3
rest GY,ST;cond=LOC.eq.i.and.ROTA.eq.k

```

```

BLOC REP/FERT1
TREA FERT1
BLOC SERIES.REP.FERT1/CYCLE
TREA FERT1*SERIES*CYCLE +SERIES.REP/CYCLE-REP.FERT1.SERIES/CYCLE
ANOV[SE=M;FPRO=Y; FACT=4] GY; FITT=F1; RESI=R1
GRAPH R1;F1
endf: endf

```

clos:stop

File LTRT14.GEN

"

This prog is prepared for training on Long-term rotational trials,
 7-18 Mar 93 conducted by FRMP, ICARDA.

File B_LVF1.TXT has data on
 3 rotations (B/V,B/F,B/B0;
 3 fertility treatments

(N20:P60 0: 0,0:0 0:0,N20:P30 N20:P30 N20:30,0:0 N20:30);
 2 sites (TelHadya, Breda),
 6 years (Y83_84,Y84_85...Y88_89).
 We generate 2 Series (Odd, Even), 3 cycles (First, Second, Third).

**** Analysis of data combined over locations or/and rotations*****

```

UNIT[ 288]
OPEN 'B_LVF1.TXT'; CH=2;FI=IN
FACT[ LEVE=2; Labels=It(TelHadya, Breda)] LOC
Fact[ leve=2; Labels=It(Odd, Even)] SERIES
FACT[LEVE=22] TREAT
FACT[ LEVE=3] REP
FACT[LEVE=6; Labe=It(Y83_84,Y84_85,Y85_86,Y86_87,Y87_88,Y88_89)] YEAR
READ [ch=2] TREAT, LOC,REP,YEAR,SERIES,GY,ST; frep=lev

FACT[LEVE=6; labe=It(B_P, B_PB, B_V, B_VB, B_F, B_B)] ROTA
CALC ROTA=NEWL(TREAT;I((1,2,3,4)4,5,5,6,6,6,6))
FACT[LEVE=3; Labe=It(First, Second, Third)] CYCLE
CALC CYCLE=NEWL(YEAR;I(1,1,2,2,3,3))
FACT[LEVE=3; Labe=It('-F','+',F_Others)] FERT1
CALC FERT1=NEWL(TREAT;I(3,3,2,3, 3,3,3,3, 3,3,3,3, 3,3,1,3, 1,2, 3,3,1,2))
dele TREAT

" FOR B/V,B/F and b/b"

" for Tel Hadya (i.e. LOC=1) only but combined over rotations B_V,B_F,B_B"

FACT[LEVE=4; labe=It(B_V,B_F,B_B, Others)] ROTA1
CALC ROTA1=NEWL(ROTA;I(4,4,1,4,2,3))
DELE ROTA

REST GY,F1,R1; COND=LOC.EQ.1.AND.ROTA1.LT.4.AND.FERT1.LT.3

BLOC SERIES.REP.FERT1.ROTA1/CYCLE
TREA FERT1*ROTA1*SERIES*CYCLE +SERIES.REP/CYCLE-REP.(FERT1*ROTA1).SERIES/CYCLE
ANOV[SE=M;FPRO=Y; FACT=4] GY; FITT=F1; RESI=R1
GRAPH R1;F1

" for only B_V rotation(i.e. ROTA1=1)but over both locations TelHadya and Breda"

REST GY,F1,R1; COND=ROTA1.eq.1.AND.FERT1.LT.3
BLOC LOC.SERIES.REP.FERT1/CYCLE
TREA LOC*FERT1*SERIES*CYCLE+REP.SERIES.LOC/CYCLE \
-REP.FERT1.SERIES.LOC/CYCLE
ANOV[SE=M;FPRO=Y;FACT=5] GY;FITT=F1;RESI=R1
GRAPH R1;F1

" Combined over all rotations and locations "
```

```

REST GY,F1,R1; COND=ROTA1.LT.4.AND.FERT1.LT.3
BLOC LOC.SERIES.REP.FERT1.ROTA1/CYCLE
TREA LOC*FERT1*ROTA1*SERIES*CYCLE+REP.SERIES.LOC/CYCLE \
      -REP.(FERT1*ROTA1).SERIES.LOC/CYCLE
ANOV[SE=M;FPRO=Y;FACT=5] GY;FITT=F1;RESI=R1
GRAPH R1;F1
clos
stop

```

File LTRT21.GEN

"

This program illustrates analysis of data on Long-term rotational trials conducted by Dr. M. Jones, FRMP and has been preprepared for training on Long-term rotational trials, 7-18 Mar 93 conducted by Dr. M. Jones,FRMP,ICARDA. The data in file HAYTDM6.txt are on hay and total barley dry matter produced from

```

2 rotations: Barley/Vetch and Barley/(Vetch+Barley)
4 fertility treatments (20:60 0:0 20:30 0:30 20:0 0:60 0:0 0:0
6 years(1983/84 ... 1988/89)
3 reps
2 sites

```

"

UNITS[288]

OPEN ' HAYTDM6.TXT' ; CH=2;FI=IN

```

FACT[ LEVE=2; Labels=lt(TelHadya, Breda)] LOC
FACT[LEVE=22] TREAT
FACT[ LEVE=3] REP
FACT[LEVE=6; Labe=lt(Y83_84,Y84_85,Y85_86,Y86_87,Y87_88,Y88_89)] YEAR
READ [CH=2] LOC,YEAR,REP,TREAT,HAY,TDM ; frep=lev

```

```

FACT[LEVE=6; labe=lt(B_P, B_PB, B_V, B_VB, B_F, B_B)] ROTA
CALC ROTA=NEWL(TREAT;l((1,2,3,4)4,5,5,6,6,6,6))
Fact[ leve=2; Labels=lt(Odd, Even)] SERIES
CALC SERIES=NEWL(YEAR;l(1,2,1,2,1,2))
FACT[LEVE=3; Labe=lt(First, Second, Third)] CYCLE
CALC CYCLE=NEWL(YEAR;l(1,1,2,2,3,3))

```

```

FACT[LEVE=5;Labe=lt('20:60 0:0','20:30 0:30','20:0 0:60','0:0 0:0',F_others)] \
      FERT1
CALC FERT1 = NEWL(TREAT;l(5,5,1,1, 5,5,2,2, 5,5,3,3, 5,5,4,4, 6(5)))

```

```

dele TREAT
PRINT LOC,REP, ROTA, YEAR, FERT1,HAY,TDM; fiel=5(13),2(6); deci=2

```

```

for i=1,2 : for j=3,4

```

```

print i,j
rest HAY,TDM; cond=LOC.eq.i.AND.ROTA.EQ.j

```

```
Tabu[class= FERT1,YEAR,SERIES,REP] HAY,TDM; TOTA=TBHAY,TBTDM
print[ miss=' ' ] TBHAY,TBTDM; field=6; deci=2
```

```
endf : endf
clos: stop
```

File LTRT22.GEN

```
" *****
This program illustrates analysis of data on Long-term rotational trials
conducted by Dr. M. Jones, FRMP and has been preprepared for training on
Long-term rotational trials, 7-18 Mar 93 conducted by Dr. M. Jones,FRMP,ICARDA.
The data in file HAYTDM6.txt are on hay and total barley dry matter produced
from
```

```
2 rotations: Barley/Vetch and Barley/(Vetch+Barley)
4 fertility treatments (20:60 0:0 20:30 0:30 20:0 0:60 0:0 0:0
6 years(1983/84 ... 1988/89)
3 reps
2 sites
```

```
**** Analysis of data for individual locations,years, rotations****
"
```

```
UNITS[288]
OPEN ' HAYTDM6.TXT' ; CH=2;FI=IN
```

```
FACT[ LEVE=2; Labels=lt(TelHadya, Breda)] LOC
FACT[LEVE=22] TREAT
FACT[ LEVE=3] REP
FACT[LEVE=6; Labe=lt(Y83_84,Y84_85,Y85_86,Y86_87,Y87_88,Y88_89)] YEAR
READ [CH=2] LOC,YEAR,REP,TREAT,HAY,TDM ; frep=lev
```

```
FACT[LEVE=6; labe=lt(B_P, B_PB, B_V, B_VB, B_F, B_B)] ROTA
CALC ROTA=NEWL(TREAT;l((1,2,3,4)4,5,5,6,6,6))
Fact[ leve=2; Labels=lt(Odd, Even)] SERIES
CALC SERIES=NEWL(YEAR;l(1,2,1,2,1,2))
FACT[LEVE=3; Labe=lt(First, Second, Third)] CYCLE
CALC CYCLE=NEWL(YEAR;l(1,1,2,2,3,3))
```

```
FACT[LEVE=5;Labe=lt('20:60 0:0','20:30 0:30','20:0 0:60','0:0 0:0',F_others)] \
FERT1
CALC FERT1 = NEWL(TREAT;l(5,5,1,1, 5,5,2,2, 5,5,3,3, 5,5,4,4, 6(5)))
```

```
dele TREAT
```

```
rest TDM;cond=LOC.eq.1.and.YEAR.eq.1.and.ROTA.eq.3
```

```
BLOC REP/FERT1
TREA FERT1
ANOV[PRINT=a,m,mi,%cv; SE=M;FPRO=Y] TDM ; FITT=F1; RESI=R1
GRAPH R1;F1
```

clos:stop

File LTRT23.GEN

"

This program illustrates analysis of data on Long-term rotational trials conducted by Dr. M. Jones, FRMP and has been preprepared for training on Long-term rotational trials, 7-18 Mar 93 conducted by Dr. M. Jones, FRMP, ICARDA. The data in file HAYTDM6.txt are on hay and total barley dry matter produced from

2 rotations: Barley/Vetch and Barley/(Vetch+Barley)
4 fertility treatments (20:60 0:0 20:30 0:30 20:0 0:60 0:0 0:0
6 years(1983/84 ... 1988/89)
3 reps
2 sites

**** Analysis of data for individual location and rotation combinations ****

"

UNITS[288]

OPEN ' HAYTDM6.TXT' ; CH=2;FI=iN

FACT[LEVE=2; Labels=lt(TelHadya, Breda)] LOC

FACT[LEVE=22] TREAT

FACT[LEVE=3] REP

FACT[LEVE=6; Labe=lt(Y83_84,Y84_85,Y85_86,Y86_87,Y87_88,Y88_89)] YEAR

READ [CH=2] LOC,YEAR,REP,TREAT,HAY,TDM ; frep=lev

FACT[LEVE=6; labe=lt(B_P, B_PB, B_V, B_VB, B_F, B_B)] ROTA

CALC ROTA=NEWL(TREAT;l((1,2,3,4)4,5,5,6,6,6,6))

Fact[leve=2; Labels=lt(Odd, Even)] SERIES

CALC SERIES=NEWL(YEAR;l(1,2,1,2,1,2))

FACT[LEVE=3; Labe=lt(First, Second, Third)] CYCLE

CALC CYCLE=NEWL(YEAR;l(1,1,2,2,3,3))

FACT[LEVE=5;Labe=lt('20:60 0:0','20:30 0:30','20:0 0:60','0:0 0:0',F_others)] \

FERT1

CALC FERT1 = NEWL(TREAT;l(5,5,1,1, 5,5,2,2, 5,5,3,3, 5,5,4,4, 6(5)))

dele TREAT

rest TDM;cond=LOC.eq.1.and.ROTA.eq.3.and.FERT1.lt.5

BLOC SERIES.REP.FERT1/CYCLE

TREA FERT1*SERIES*CYCLE +SERIES.REP/CYCLE-REP.FERT1.SERIES/CYCLE

ANOV[SE=M;FPRO=Y] TDM ; FITT=F1; RESI=R1

GRAPH R1;F1

clos:stop

File LTRT24.GEN

"

This program illustrates analysis of data on Long-term rotational trials conducted by Dr. M. Jones, FRMP and has been prepared for training on Long-term rotational trials, 7-18 Mar 93 conducted by Dr. M. Jones, FRMP, ICARDA. The data in file HAYTDM6.txt are on hay and total barley dry matter produced from

2 rotations: Barley/Vetch and Barley/(Vetch+Barley)
4 fertility treatments (20:60 0:0 20:30 0:30 20:0 0:60 0:0 0:0)
6 years(1983/84 ... 1988/89)
3 reps
2 sites

**** Analysis of data combined over location or/and rotations ****

"

UNITS[288]

OPEN ' HAYTDM6.TXT' ; CH=2;FI=IN

FACT[LEVE=2; Labels=lt(TelHadya, Breda)] LOC

FACT[LEVE=22] TREAT

FACT[LEVE=3] REP

FACT[LEVE=6; Labe=lt(Y83_84,Y84_85,Y85_86,Y86_87,Y87_88,Y88_89)] YEAR

READ [CH=2] LOC,YEAR,REP,TREAT,HAY,TDM ; frep=lev

FACT[LEVE=6; labe=lt(B_P, B_PB, B_V, B_VB, B_F, B_B)] ROTA

CALC ROTA=NEWL(TREAT;l((1,2,3,4)4,5,5,6,6,6,6))

Fact[leve=2; Labels=lt(Odd, Even)] SERIES

CALC SERIES=NEWL(YEAR;l(1,2,1,2,1,2))

FACT[LEVE=3; Labe=lt(First, Second, Third)] CYCLE

CALC CYCLE=NEWL(YEAR;l(1,1,2,2,3,3))

FACT[LEVE=5;Labe=lt('20:60 0:0','20:30 0:30','20:0 0:60','0:0 0:0',F_others)] \

FERT1

CALC FERT1 = NEWL(TREAT;l(5,5,1,1, 5,5,2,2, 5,5,3,3, 5,5,4,4, 6(5)))

dele TREAT

" **** For Telhadya(LOC=1) but combined over B/V and B/BV **** "

rest TDM;cond=LOC.eq.1.and.(ROTA.eq.3.or.ROTA.eq.4).and.FERT1.lt.5

BLOC SERIES.REP.FERT1.ROTA/CYCLE

TREA FERT1*ROTA*SERIES*CYCLE +SERIES.REP/CYCLE-REP.(FERT1*ROTA).SERIES/CYCLE

ANOV[SE=M;FPRO=Y; FACT=4] TDM ; FITT=F1; RESI=R1

GRAPH R1;F1

" **** For B/V but combined over both locations **** "

rest TDM;cond=ROTA.eq.3.and.FERT1.lt.5

BLOC LOC.SERIES.REP.FERT1/CYCLE

```
TREA LOC*FERT1*SERIES*CYCLE+REP.SERIES.LOC/CYCLE-REP.FERT1.SERIES.LOC/CYCLE
ANOV[SE=M;FPRO=Y; FACT=4] TDM ; FITT=F1; RESI=R1
GRAPH R1;F1
```

" **** Combined over locations and B/V and B/BV ***** "

```
rest TDM;cond=(ROTA.eq.3.or.ROTA.eq.4).and.FERT1.lt.5
BLOC LOC.SERIES.REP.FERT1.ROTA/CYCLE
TREA LOC*FERT1*ROTA*SERIES*CYCLE+REP.SERIES.LOC/CYCLE \
-REP.(FERT1*ROTA).SERIES.LOC/CYCLE
ANOV[SE=M;FPRO=Y; FACT=5] TDM ; FITT=F1; RESI=R1
GRAPH R1;F1
```

clos: stop

File LTRT31.GEN

"

This prog is prepared for training on Long-term rotational trials,
7-18 Mar 93 conducted by FRMP, ICARDA.

File B_LVF1.TXT has data on continuous barley (B/B); four fertility
treatments (N20:P60 N20:60, N0:P0 N20:P60 N20:P60 N0:P0, N0:P0,N0:P0);
two sites (TelHadya, Breda),
6 years (Y83_84,Y84_85...Y88_89).

"

```
UNIT[ 288]
OPEN 'B_LVF1.TXT'; CH=2;FI=IN
FACT[ LEVE=2; Labels=lt(TelHadya, Breda)] LOC
FACT[LEVE=22] TREAT
FACT[leve=2] SERIES
FACT[ LEVE=3] REP
FACT[LEVE=6; Labe=lt(Y83_84,Y84_85,Y85_86,Y86_87,Y87_88,Y88_89)] YEAR
READ [ch=2] TREAT, LOC,REP,YEAR,SERIES,GY,ST; frep=lev
```

```
FACT[LEVE=6; labe=lt(B_P, B_PB, B_V, B_VB, B_F, B_B)] ROTA
CALC ROTA=NEWL(TREAT;l((1,2,3,4)4,5,5,6,6,6,6))
```

```
FACT[LEVE=5;Labe=lt('20:60 20:60','20:60 0:0','0:0 0:0','0:0 20:60',F_Others)]\
FERT
CALC FERT=NEWL(TREAT;l(18(5),1,2,3,4))
```

dele TREAT, SERIES

" FOR B/B "

```
FOR I=1...2 :FOR J=1...3 : FOR K=1...4
REST GY,ST; COND=ROTA.EQ.6.AND.LOC.EQ.I.AND.REP.EQ.J.AND.FERT.EQ.K
PRINT LOC,REP, FERT,YEAR,GY,ST; fiel=4(13),2(6); deci=2
```

ENDF: ENDF: ENDF

REST GY,ST; COND=ROTA.EQ.6

Tabu[class= LOC,FERT, YEAR,REP] GY,ST; TOTA=TBGY,TBSY

print[miss=' '] TBGY,TBSY; field=6; deci=2

FACT[LEVE=5] FERTX

CALC FERTX=FERT

FOR I=1...2 :FOR J=1...3 : FOR K=1...4

REST GY,ST; COND=ROTA.EQ.6.AND.LOC.EQ.I.AND.REP.EQ.J.AND.FERTX.EQ.K

PRINT[RL=INT;CL=INT] LOC,REP, FERTX,YEAR,GY,ST; fiel=4(13),2(6); deci=2

ENDF: ENDF: ENDF

clos: stop

File LTRT32.GEN

"

This prog is prepared for training on Long-term rotational trials,
7-18 Mar 93 conducted by FRMP, ICARDA.

File B_LVF1.TXT has data on continuous barley (B/B); four fertility
treatments (N20:P60 N20:60, N0:P0 N20:P60 N20:P60 N0:P0, N0:P0,N0:P0);
two sites (TelHadya, Breda),
6 years (Y83_84,Y84_85...Y88_89).

***** Individual locations and polynomial contrasts over years

"

UNIT[288]

OPEN 'B_LVF1.TXT'; CH=2;FI=IN

FACT[LEVE=2; Labels=It(TelHadya, Breda)] LOC

FACT[LEVE=22] TREAT

FACT[leve=2] SERIES

FACT[LEVE=3] REP

FACT[LEVE=6; Labe=It(Y83_84,Y84_85,Y85_86,Y86_87,Y87_88,Y88_89)] YEAR

READ [ch=2] TREAT, LOC,REP,YEAR,SERIES,GY,ST; frep=lev

FACT[LEVE=6; labe=It(B_P, B_PB, B_V, B_VB, B_F, B_B)] ROTA

CALC ROTA=NEWL(TREAT;I((1,2,3,4)4,5,5,6,6,6,6))

FACT[LEVE=5;Labe=It('20:60 20:60','20:60 0:0','0:0 0:0','0:0 20:60',F_Others)]\n
FERT

CALC FERT=NEWL(TREAT;I((18(5),1,2,3,4))

dele TREAT, SERIES

" FOR B/B "

REST GY; COND=ROTA.EQ.6

REST GY ; COND=LOC.EQ.1.AND.ROTA.eq.6.and.FERT.LT.5

BLOC REP/(YEAR*FERT)

TREA YEAR*FERT

ANOV[SE=M;FPRO=Y] GY ; FITT=F1 ;RESI=R1

GRAPH R1;F1

BLOC

TREA POL(YEAR;4)*REP*FERT

ANOV[PRINT=A] GY

clos : stop

File LTRT33.GEN

"

This prog is prepared for training on Long-term rotational trials,
7-18 Mar 93 conducted by FRMP, ICARDA.

File B_LVF1.TXT has data on continuous barley (B/B); four fertility
treatments (N20:P60 N20:60, N0:P0 N20:P60 N20:P60 N0:P0, N0:P0,N0:P0);
two sites (TelHadya, Breda),
6 years (Y83_84,Y84_85...Y88_89).

***** Combined over locations and polynomial contrasts over years

"

UNIT[288]

OPEN 'B_LVF1.TXT'; CH=2;FI=IN

FACT[LEVE=2; Labels=lt(TelHadya, Breda)] LOC

FACT[LEVE=22] TREAT

FACT[leve=2] SERIES

FACT[LEVE=3] REP

FACT[LEVE=6; Labe=lt(Y83_84,Y84_85,Y85_86,Y86_87,Y87_88,Y88_89)] YEAR

READ [ch=2] TREAT, LOC,REP,YEAR,SERIES,GY,ST; frep=lev

FACT[LEVE=6; labe=lt(B_P, B_PB, B_V, B_VB, B_F, B_B)] ROTA

CALC ROTA=NEWL(TREAT;l((1,2,3,4)4,5,5,6,6,6,6))

FACT[LEVE=5;Labe=lt('20:60 20:60','20:60 0:0','0:0 0:0','0:0 20:60',F_Others)]\n
FERT

CALC FERT=NEWL(TREAT;l(18(5),1,2,3,4))

dele TREAT, SERIES

" FOR B/B "

REST GY; COND=ROTA.EQ.6

REST GY ; COND=ROTA.eq.6.and.FERT.LT.5

```
BLOC LOC/REP/(YEAR*FERT)
TREA YEAR*LOC*FERT
ANOV[SE=M;FPRO=Y] GY ; FITT=F1;RESI=R1
GRAPH R1;F1
```

```
BLOC
TREA POL(YEAR;5)*LOC*REP*FERT
ANOV[PRINT=A;FACT=4] GY
```

```
clos : stop
```

```
File LTRT34.GEN
```

```
" .....
```

This prog is prepared for training on Long-term rotational trials,
7-18 Mar 93 conducted by FRMP, ICARDA.

File CONT_BAR.TXT has data on continuous barley (B/B); four fertility
treatments (N20:P60 N20:60, N0:P0 N20:P60 N20:P60 N0:P0, N0:P0,N0:P0);
two sites (TelHadya, Breda),
6 years (Y83_84,Y84_85...Y88_89).

***** Polynomial contrast computed for analysis

```
"
```

```
UNIT[ 6]
OPEN 'CONT_BAR.TXT'; CH=2;FI=IN
VARI[VALU= -5,-3,-1,1,3,5] Linear
VARI[VALU= 5, -1,-4,-4,-1,5] Quadra
VARI[VALU= -5, 7, 4, -4, -7, 5]Cubic
VARI[VALU= 1, -3, 2, 2, -3, 1] Quartic
VARI[VALU= -1, 5, -10, 10, -5, 1] Quintic
VARI[NVAL=24] GLin,GQua,GCub,GQuar,GQui,SLin,SQua,SCub,SQuar,SQui
```

```
FOR I=1...24
READ[CH=2; END=*; FORM=I(-4,2)] GY,ST
```

```
CALC GLin $[I] = SUM(GY*Linear) : CALC GQua $[I] =SUM(GY*Quadra)
CALC GCub $[I] = SUM(GY*Cubic) : CALC GQuar $[I]= SUM(GY*Quartic)
CALC GQui $[I] = SUM(GY*Quintic)
```

```
CALC SLin $[I] =SUM(ST*Linear) : CALC SQua $[I] =SUM(ST*Quadra)
CALC SCub $[I] =SUM(ST*Cubic) : CALC SQuar $[I]=SUM(ST*Quartic)
CALC SQui $[I] =SUM(ST*Quintic)
```

```
endf
```

```
FACT[ LEVE=2; Labels=lt(TelHadya, Breda); values=12(1,2)] LOC
FACT[ LEVE=3; values=4(1...3)2] REP
FACT[leve=4; Labe=lt('20:60 20:60','20:60 0:0','0:0 0:0','0:0 20:60') ; \
valu=(1...4)6] FERT
```

```
BLOC LOC/REP/FERT
TREA LOC*FERT
FOR Y=GLin,GQua,GCub,GQuar,GQui
ANOV[SE=M;FPRO=Y] Y
endf
```

clos : stop

```
poly n=6
lin = -5,-3,-1,1,3,5
quadr = 5, -1,-4,-4,-1,5
cubic = -5 7 4 -4 -7 5
quart = 1 -3 2 2 -3 1
quint = -1 5 -10 10 -5 1
```

Outputs : Not included

**Computer and Biometrics Services Unit
ICARDA**

Serial Number -007

GENSTAT Modules in Computing Biometrics

Title: Modelling germination in wheat- A robust method

1. Introduction:

This program models seed germination with temperature and times. For each genotype and temperature combination in the experiment, it fits a logistic model to the cumulative percentage germination as a function of time, computes time (D50%) to 50% germination, computes rate (R50%=1/D50) of germination as inverse of this time. The relationship between D50% and temperature was roughly appearing like a quadratic divided by linear (qdl) function in temperature for each of the genotypes. Estimation of this function is essential to obtain optimum temperature and hence the base temperature. However, the fit was not very satisfactory when all the data points were used. Scatter and fitted plots were examined to notice the presence of few suspected outliers (suspected by high standardized residuals). Then, we decided to fit the model again without such points {i.e. the points where standardized residuals were outside (-2,2) interval}, and the suspected outliers were identified again. This process was iterated and was found to converge within five iterations (convergence in terms of residual mean square, values of the estimates of qdl parameters, and the suspected outliers whose numbers varied from 1 to 5 on an average of 50 sample points). The experience was similar with each of the 18 genotypes. With a loss of few points or degree of freedom, the gain was enormous (minimum increase in %R-square(adj) exceeded 20%). The linear function in times are fitted to R50% for cases below the optimum temperature. The base temperature is estimated with standard error.

2. Areas of application and keywords:

Plant Physiology, abiotic stress, temperature stress, base temperature, germination, wheat

3. Input/Output:

Input:

For a number of genotype and temperature, number of seeds germinated, total number of seeds used for different times in hours.

Output:

Estimates of base temperature, optimum temperature, R50%, D50%

4. Location of the program:

Program:

A:\CompBiom\94\Germ_Rob.gen
(Software: GENSTAT 5)

5. Location of illustrative data and results:

Data:

ICARDA VAX [murari.CP.maha.assad]Ger1.dat, Ger2.dat, Temp1.mea, Temp2.mea

Results:

ICARDA VAX [murari.CP.maha]Germ_Rob.opt

6. Client(s)

Dr. V. Mahalakshmi, CP

7. Date

June 1994

8. Comments

M. Singh

Program:

" Program 2: Lag phase "

```

Unit[864]
Open 'coeff12.opt'; ch=2
Fact[leve=18]Geno
Read[ch=2] Cases,Geno,Temp,Rsq,BEst,MEst,CEst,AEst,D50
Vari Indicate
Calc Indicate=0

Vari[nval=18]Msq,Rsq, Df, OpTemp, BaseT,SeBaseT,Slope,SeSlope

For i=1...18
rest Lag,MEst,BEst,CEst,AEst,Temp ; cond=Geno.eq.i.and.D50.gt.0
Calc Lag=MEst-(1/BEst)*Log(-1-CEst/AEst)

rest Temp,f,Lag,res ; cond=Geno.eq.i.and.D50.gt.0

Model Lag;fitt=f
Fitc[prin='] Temp
Rkeep Resi=res

Calc res=abs(res)

For[ntimes=5]
Rest Temp,Lag,f,res
rest Temp,f,res,Indicate, Lag ; cond=Geno.eq.i.and.D50.gt.0.and.res.le.2
Model Lag;fitt=f
Fitc[prin='] Temp
Rkeep Resi=res
calc res=abs(res)
Endf
Calc Indicate=Indicate+1

Model Lag;Fitted=Fval
Fitc Temp
graph Lag,Fval;Temp
rest Lag,MEst,BEst,CEst,AEst,Temp

endf
Rest Indicate

Rest Lag, Temp, Geno ; Indicate.gt.0

Model Lag
Terms Temp+Geno+Geno.Temp
Fitc[prin=m,s; Fprob=y] Temp
Add [prin=m,s; Fprob=y]Geno
Add [prin=m,s;Fprob=y] Geno.Temp

```

Add[print=m,s,a,e; nonl=s; Fprob=y;Tprob=y]

clos: stop

clos
stop

" Program 1: Base temperature"

Unit[864]

Open 'coeff12.opt'; ch=2

Fact[leve=18]Geno

Read[ch=2] Cases,Geno,Temp,Rsq,B1,M1,C1,A1,D50

Scal A,B,C,D, TOPT,df,ss

Var[nval=18]Msq,Rsq, Df, OpTemp, BaseT,SeBaseT,Slope,SeSlope

Device 5

Pen 1, 2 ; meth=p,m

For i=1...18

rest Temp,f,res, D50,R50 ; cond=Geno.eq.i.and.D50.gt.0

Calc R50=1/D50

Model D50;fitt=f

Fitc[curve=qdl ; prin=*] Temp

Rkeep Resi=res

"Dgraph D50,f; Temp ; pen=1,2"

Calc res=abs(res)

For[ntimes=5]

Rest Temp,D50,f,res

rest Temp,D50,f,res ; cond=Geno.eq.i.and.D50.gt.0.and.res.le.2

Model D50;fitt=f

Fitc[curve=qdl;prin=*;fprob=y] Temp

Rkeep Resi=res

calc res=abs(res)

Endf

"Dgraph D50,f; Temp ; pen=1,2"

Rkeep df=df; devi=ss ; Est=Est

Calc Msq[i]=ss/df : & Rsq[i]=(1-ss/df/var(D50))*100. :& Df[i]=df

CALC D,B,C,A=Est\${1...4}

Calc TOPT=(sqrt(B*D/C)-1)/D

Calc OpTemp[i]=TOPT

Rest R50,Temp,D50,f,res

```
Rest R50,Temp; Geno.eq.i.and.D50.gt.0.and.res.le.2.and.Temp.le.TOPT
Model R50
```

```
Fit[prin='] Temp
Rkeep Est=Est1 ; Vcov=Vcov
Scal Btemp
Calc Btemp=-Est1$[1]/Est1$[2]
```

```
Calc BaseT$[i] =Btemp
Calc Slope$[i] =Est1$[2]
Calc SeSlope$[i]=Sqrt(Vcov$[2;2])
Calc SeBaseT$[i]=ABS(Btemp)*SQRT(Vcov$[1;1]/(Est1$[1])**2+ \
    Vcov$[2;2]/(Est1$[2])**2-2.*Vcov$[2;1]/Est1$[1]/Est1$[2])
```

```
Rest R50,D50,Temp
```

```
Endf
```

```
Fact[leve=18; valu=1...18] Genotype
Calc Slope=1000*Slope :& SeSlope=1000*SeSlope
```

```
Print Genotype, Rsq, OpTemp,BaseT,SeBaseT,Slope,SeSlope ;fiel=9
Scal Chsq,PrChi,wmean,Sewmean
Calc wmean=sum(BaseT/SeBaseT/SeBaseT)/Sum(1/SeBaseT/SeBaseT)
Calc Chsq=Sum(((BaseT-wmean)/SeBaseT)**2)
Calc PrChi=1-Chsq(Chsq;17)
Calc Sewmean=sqrt(1/Sum(1/SeBaseT/SeBaseT))
Prin Chsq,PrChi,wmean,Sewmean
```

```
Calc wmean=sum(Slope/SeSlope/SeSlope)/Sum(1/SeSlope/SeSlope)
Calc Chsq=Sum(((Slope-wmean)/SeSlope)**2)
Calc PrChi=1-Chsq(Chsq;17)
Prin Chsq,PrChi
```

```
Rest BaseT,SeBaseT; cond=BaseT.ge.0
Calc wmean=sum(BaseT/SeBaseT/SeBaseT)/Sum(1/SeBaseT/SeBaseT)
Calc Chsq=Sum(((BaseT-wmean)/SeBaseT)**2)
Calc PrChi=1-Chsq(Chsq;14)
Calc Sewmean=sqrt(1/Sum(1/SeBaseT/SeBaseT))
Prin Chsq,PrChi,wmean,Sewmean
```

```
clos
stop
```

```
Unit[48]
Open 'Ger2.dat'; ch=2 ; width=320
Open 'Temp2.mea';ch=3
```

```

Scal Rep,Rep1,Temp,Temp1,Cham, Cham1,TMeas, Geno
Scal Temp,Geno, B,M,C,A, Tot, Rbarsq
Scal[valu=*] Bmis,Mmis,Cmis,Amis

```

```

Read[ch=2; end=*) Time
Var[nval=52]Fullrow

```

```

Open 'coeff2.opt';ch=4;fi=out

```

```

For[ntimes=72]
Read[ch=3;end=*) Rep1,Temp1,Cham1,TMeas
For[ntimes=6]
Read[ch=2;end=*) Fullrow

```

```

Calc Temp, Cham,Rep,Geno=Fullrow${1...4}
Calc Y=Fullrow${1(5...52)}
Calc Tot=Sum(Y)

```

```

If Tot.le.12
  Calc D50,Rbarsq=0,0 : & B,M,C,A=Bmis,Mmis,Cmis,Amis

```

```

Print[ch=4;ipri=*) Geno,TMeas,Rbarsq,B,M,C,A,D50;fiel=4,2(6),5(9);deci=2,2,5(3)
Else

```

```

Calc Y=Cum(Y) : Calc Y=(Y/Tot)*100.
Model Y ; Fitt=F
Fitc[ curve=log] Time
Graph[ncol=35;nrow=16] Y, F;Time; Meth=p,c
Rkeep Est=Est ;df=df;devi=ss
Calc Rbarsq=100.*(1-ss/df/var(Y))
Equa Est; lp(B,M,C,A)
Calc D50=M-Log((C+A-50)/(50-A))/B

```

```

Print[ch=4; ipri=*) Geno,TMeas,Rbarsq,B,M,C,A,D50;fiel=4,2(6),5(9);deci=2,2,5(3)

```

```

Endif

```

```

Endf
Endf

```

```

clos
stop

```

Note: Coeff12.opt is file (Coeff1.opt +Coeff2.opt) combined over
experiments 1 and 2

**Computer and Biometrics Services Unit
ICARDA**

Serial Number -008

GENSTAT Modules in Computing Biometrics

Title: Modelling absorbance due to rhizobia under salinity

1. Introduction:

This program models absorbance percentage in various rhizobia strain with time under two salinity levels. For each rhizobium and salinity level combination in the experiment, it fits a logistic model to the cumulative percentage germination as a function of time, computes lag phase, maximum growth, slope (at point of inflection). These parameters are then subjected to regression on rhizobia and salinity leading to their predictions.

2. Areas of application and keywords:

Plant Physiology, abiotic stress, biotic stress, chickpea

3. Input/Output:

Input:

For a number of rhizobia, salinity, absorption %, over different times in hours.

Output:

Estimates of lag phase, slope, maximum growth rate. Predictions.

4. Location of the program:

Program:

A:\CompBiom\94\Absorb.gen
(Software: GENSTAT 5)

5. Location of illustrative data and results:

Data:

ICARDA VAX [murari.LP.MCS.Adlan]Absorb.txt

Results:

ICARDA VAX [murari.LP.MCS.ADLAN]Absorb.out

6. Client(s)

Mr. Adlan, (M.Sc. student at AUB, Lebanon)

7. Date

September 1994

8. Comments

M. Singh

Program

Data on response of rhizobia to salinity in terms of absorbance over a period of seven days. 18 rhizobia, 2 salinity levels.
 Lab expt. conducted at T.Hadya during August 1993 by Mr. Adlan, ARC, Sudan.
 This work is for his M.Sc. degree. Supervisor: Dr. M. C. Saxena.

Time is in hours. Asorbence in Absorbance unit

```
Unit[72]
Vari[nval=14]Absorb,Time: & [nvalu=17] Fullrow
Open 'Adlanr.yyy'; ch=2;fi=in
Skip[ch=2]1
Scal Ncases; 72
Scal SRhiz,SSalin,SSampl, SB,SM,SC,SA,SRbar, AbsLast
Scal[valu=*] Bmis,Mmis,Cmis,Amis, Miss
Scal Two; 2
Vari[valu=0,4,22,28,46,52,70,76,94,100,118,124,142,148] Time
Fact[leve=18]Rhiz: &[leve=2]Salin,Sampl
Vari B,M,C,A,Rbarsq
Fact[leve=2;valu=72(1); labe=!(Responsive,Constant) } ID
For i=1...Ncases

Read[ch=2;end=*] Fullrow
Calc (SRhiz,SSalin,SSampl)=Fullrow$[1...3] : & Absorb=Fullrow$[1(4...17)]
  If NOBS(Absorb).eq.0
    Calc SB,SM,SC,SA,SRbar=Miss
    Calc (B,M,C,A)$[i]=SB,SM,SC,SA
    Calc Rbarsq[i]=SRbar
    Calc ID[i]=Int(Two)
    Print SRhiz,SSalin,SSampl
    Calc (Rhiz,Salin,Sampl)$[i]=Int(SRhiz,SSalin,SSampl)
  else

Rest Absorb; Time.ne.0.
Model Absorb ; Fitt=F
Fitc[ curve=log; print=*] Time
Rest Absorb

Rkeep Est=Est ;df=df;devi=ss
Calc SRbar=100.*(1-ss/df/var(Absorb))
Calc Rbarsq[i]=SRbar
Calc (Rhiz,Salin,Sampl)$[i]=Int(SRhiz,SSalin,SSampl)
Equa Est; lp(SB,SM,SC,SA)
Calc (B,M,C,A)$[i]=SB,SM,SC,SA
"Graph[ncol=35;nrow=16] Absorb, F;Time; Meth=p,c"
endif

endif
```

" Time90: Time to Gamma=90% of (A+C): maximum growth"

Calc Slop=B*C/4 : & Lag=M-(1/B)*Log(-1-C/A): & MaxGr=A+C

Scal Gamma ;0.90

Calc Time90=M-(1/B)*Log(-1+C/((A+C)*Gamma-A))

Calc Lag16=M-(1/B)*Log(-1+C/(0.016-A))

Calc Lag16=Lag16*(Lag16.ge.0)

Open 'adlan1.opt'; ch=2;fi=ou

Outp 2

Vari [valu=72(1)] Sel

Calc Sel=Sel+(Rhiz.eq.5.and.Salin.eq.2)+(Rhiz.eq.18.and.Salin.eq.2)

Calc Lag1=Lag*(Lag.ge.0)

For y=Slop,Lag,Lag1,Lag16,MaxGr,Time90 "B,M,C,A"

Model y

Rest y; Sel.eq.1

Fit[prin=m,s,a; fprob=y] Rhiz+Salin+Rhiz*Salin

Rest y

Print ' ***** Warning: do not extract estimates for rhiz,sal=(5,2), (18,2)'

Pred[prin=p,s; alias=ignore;comb=p]Rhiz

Pred[prin=p,s;alias=ignore;comb=p]Salin

Pred[prin=p,s;alias=ignore;comb=p]Salin,Rhiz

endf

Print Rhiz,Salin,Sampl,ID,Rbarsq, M,Slop,Lag,MaxGr,Time90; Fiel=8

clos :stop

Output : Not included